Working with the IFS in RPG IV

Scott Klement

Working with the IFS in RPG IV

by Scott Klement

This eBook is intended to help an experienced RPG IV programmer learn how to read, write and manipulate documents within the Integrated File System on an IBM iSeries/400 server.

It is assumed that the reader of this tutorial is already familiar with the RPG IV language, including prototypes sub-procedures and service programs.

Trademarks

The terms RPG IV, RPG/400, Integrated Language Environment, C/400, OS/400, AS/400, and iSeries/400 are trademarks of International Business Machines Corporation in the United States, other countries, or both. Microsoft, MS-DOS, Windows, and Windows NT are registered trademarks of Microsoft Corporation.

UNIX is a registered trademark in the United States and other countries of X/Open Company Limited

Other company, product and/or service names may be trademarks and/or service marks of others.

Table of Contents

| 1. Introduction to the IFS | 1 |
|---|----|
| 1.1. What is the Integrated File System? | 1 |
| 1.2. What is a stream file? | |
| 1.3. What different file systems can I work with in RPG? | |
| 1.4. IFS information in the Information Center | 2 |
| 1.5. An IFS "Hello World" Application | |
| 1.6. Looking at our example from OS/400 | 4 |
| 2. The Basics of Stream Files | 7 |
| 2.1. Opening Files with the open() API | 7 |
| 2.1.1. Creating a header member | 7 |
| 2.1.2. Prototyping the open() API | 7 |
| 2.1.3. The path parameter | 8 |
| 2.1.4. The oflag parameter | 9 |
| 2.1.5. The mode parameter | 10 |
| 2.1.6. The codepage parameter | 11 |
| 2.1.7. The return value of the open() API | 11 |
| 2.1.8. Code snippet showing the use of the open() API | |
| 2.2. Closing a file with the close() API | 11 |
| 2.3. Writing streams with the write() API | |
| 2.4. Reading a stream file with the read() API | |
| 2.5. Example of writing and reading data to a stream file | |
| 2.6. Error handling | |
| 2.6.1. Retrieving the error number | |
| 2.6.2. What does the error number mean? | |
| 2.6.3. Getting a human-readable error message | |
| 2.6.4. Utilities for communicating errors | |
| 2.7. Our last example with error handling added | |
| 2.8. Example of writing raw data to a stream file | |
| 3. Other simple, but helpful IFS commands | |
| 3.1. Checking existence and permissions to files | |
| 3.2. Example of checking for an object in the IFS | |
| 3.3. Changing permissions on an existing IFS Object | |
| 3.4. Example of changing an IFS objects permissions | |
| 3.5. Retrieving Stream File Stats | |
| 3.6. Adding a *SAME option to the permission changer | |
| 3.7. Deleting IFS objects | |
| 3.8. Renaming IFS objects | |
| 3.9. Example of renaming and deleting IFS objects | |
| 4. Accessing stream files randomly | |
| 4.1. Positioning to a given point in the file | |
| 4.2. Example of using lseek() to jump around the file | |
| 4.3. Organizing a stream file into records | |
| 4.4. Calculating number of records in a file | |
| 4.5. Example of reading/writing/updating records in a stream file | 43 |

| 5. Text files | 47 |
|---|----|
| 5.1. How do text files work? | 47 |
| 5.2. Writing text data to a stream file | 47 |
| 5.3. Reading text data from a stream file | |
| 5.4. Example of writing and reading text files | 51 |
| 5.5. Using code pages with text files | 54 |
| 5.6. Example of writing & creating an ASCII text file | 55 |
| 5.7. Example of a report in ASCII | |
| 6. Additional Text Formats | |
| 6.1. Comma Separated Values (CSV) Format | |
| 6.2. Example of creating a CSV file | |
| 6.3. HTML (web page) format | 65 |
| 6.4. Example of creating an HTML file | 65 |
| 7. Working with directories | 71 |
| 7.1. How directories work | 71 |
| 7.2. Creating directories | 71 |
| 7.3. Removing directories | |
| 7.4. Switching your current directory | |
| 7.5. Opening Directories | 73 |
| 7.6. Reading Directories | 74 |
| 7.7. Closing an open directory | |
| 7.8. Example of reading a directory | |
| 7.9. Example of making a DIR command for QSHELL | 77 |
| 7.10. Example of Reading a directory recursively | |

Chapter 1. Introduction to the IFS

The purpose of this book is to teach you how to work with stream files in the Integrated File System from an ILE RPG/400 (RPG IV) program.

1.1. What is the Integrated File System?

Traditionally, we've worked with a file system on OS/400 that was made up of libraries. Within each library are objects that are assigned a specific "object type" such as a file, a program or a command. Each object type has a strictly defined layout. Files, for example, contain members, which then contain records, which contain fields. Each of these pieces is given a strict definition of what it is, how it works, and how it can be used.

By contrast, other operating systems, such as UNIX, MS-DOS and Windows use file systems where each object is simply a collection of bytes. Applications can be written to write and read these bytes as data files, but they can also view them as programs, pictures, sounds, video files, or anything else that a programmer can dream up. In other words, their contents are not strictly defined by the operating system.

At some point in it's history it was decided that OS/400 should be extended to work with these "stream files."

Some problems needed to be solved in order to do this, however, because although these file systems are all similar, they are not exactly the same. MS-DOS filenames can be 8 characters long with a 3-character "extension", and cannot contain spaces in the filename. Windows extends the MS-DOS capability by adding the ability to have a much longer file name, plus they now allow spaces. And UNIX allows spaces and long filenames, and even makes the distinction between upper & lower case letters. In other words, in Windows "MyFile.txt" and "myfile.txt" would refer to the same file, but in UNIX they refer to two different files.

In order to make OS/400 capable of working with files and folders that adhere to all of these different rules, the Integrated File System (IFS) was born. In the IFS, many different file systems can be accessed using a common interface. Special directory names are used to denote which file system you're referring to. You can even define your own file system that uses your own user-defined rules if you wish! (But, we won't be covering that in this book.)

1.2. What is a stream file?

With the exception of the original "library" file system, most of the file systems in the IFS store their objects in the "loosely-defined" file structure that's common in Windows and UNIX environments. This type of object is known as a "stream file" because the data in it is thought of as "one continuous stream of bytes." In other words, there's a start of the file, and an end to the file, but nothing else is defined. A program can use this big, long string of bytes for any purpose that it likes.

1.3. What different file systems can I work with in RPG?

There are many file systems defined to work in the IFS. Here we will just list a few of them, so that you get the idea:

| File system | Description | Works Like |
|-------------|---------------------------------------|------------|
| /QSYS.LIB | The traditional "Library file system" | OS/400 |

| File system | Description | Works Like |
|-------------|--|------------|
| /QDLS | The "Document Library Services" (OfficeVision) file system | MS-DOS |
| /QOpenSys | The "Open Systems" file system | UNIX |
| / ("root") | The "root" file system | Windows |

Most of the file systems not mentioned here are for accessing data that is not stored on your AS/400's hard drive, but is accessed on optical media, or on another computer somewhere over your LAN. Rest assured that all of these file systems can be accessed from an RPG program using the same Integrated File System interfaces that we use in this book.

1.4. IFS information in the Information Center

IBM provides a series of Application Program Interfaces (APIs) which we will use to access the IFS. These APIs are designed to be compatible with those used in UNIX environments.

Unfortunately, since most UNIX programming is done in the C programming language, almost all of the documentation assumes that you are a C programmer.

Throughout this book, we will be looking at the C documentation, and translating it into RPG for our purposes. We will make our RPG implementation as much like the C implementation as possible, so that it will be relatively easy to use the IBM manuals to find the information you're looking for in the future.

Don't worry if you're not a C programmer! In this eBook, we will explain each API in RPG terms, plus sample programs will be provided that you can use as a guide.

For general information about the Integrated File System, such as the concepts behind it, follow these steps:

- 1. Open up your Information Center CD, or point your web browser at: http://publib.boulder.ibm.com/pubs/html/as400/infocenter.htm
- 2. If asked, choose the area of the world you live in, the language that you speak, and the release of OS/400 that you need.
- 3. Using the navigator bar on the left-hand side of the screen, click "Database and File Systems", then "File Systems and Management", then "Integrated File Systems Concepts."

Tip: If your web browser has trouble with the navigation bar, (and many do) you can get to the same place by clicking on the "Site Map" instead.

For reference information on the APIs themselves, follow these:

- 1. Open up your Information Center CD, or point your web browser at: http://publib.boulder.ibm.com/pubs/html/as400/infocenter.htm
- 2. If asked, choose the area of the world you live in, the language that you speak, and the release of OS/400 that you need.

3. Using the navigator bar on the left-hand side of the screen, click "Programming", then "CL and APIs", then "OS/400 APIs", then "APIs by category".

Tip: If your web browser has trouble with the navigation bar, (and many do) you can get to the same place by clicking on the "Site Map" instead.

4. From the list of categories, click on "UNIX-Type", and then click on the "Integrated File System APIs" topic.

1.5. An IFS "Hello World" Application

I don't know about you, but I'm falling asleep reading this book. I'm not a historian, I'm a programmer! Lets go! Let's write some code, already!

Type the following code into a source member called CH1HELLO, and we'll try to make it work:

```
** HELLO WORLD in the IFS Example:
   (From Chapter 1)
 * *
 * *
 ** To compile:
       CRTBNDRPG CH1HELLO SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
 * *
 * *
H DFTACTGRP(*NO) ACTGRP(*NEW)
 ** API call to open a stream file
* *
               PR
                               10I 0 extproc('open')
D open

    value options(*string)

D path
D
   oflag
                               10I 0 value
   mode
                               10U 0 value options(*nopass)
D
                              10U 0 value options(*nopass)
D
   codepage
D O_WRONLY
                 С
                                     2
D O_CREAT
                 С
                                     8
D O_TRUNC
                 С
                                     64
                 С
D RW
                                     6
DR
                 С
                                     4
                 С
D OWNER
                                     64
D GROUP
                 С
                                     8
 ** API call to write data to a stream file
* *
D write
                PR
                              10I 0 extproc('write')
                               10I 0 value
D fildes
D buf
                                 * value
                              10U 0 value
D nbyte
```

```
** API call to close a stream file
* *
D close PR 10I 0 extproc('close')
D fildes
                           10I 0 value
             S
S
                           10T 0
D fd
D data
                           12A
D msq
              S
                            52A
C* Create an empty file called 'helloworld':
                 eval fd = open('/helloworld':
С
С
                                 O_CREAT+O_TRUNC+O_WRONLY:
                                  RW*OWNER + RW*GROUP + R )
C
                 if fd < 0
eval Msg = 'open failed!'
dsply msg
С
С
С
                 eval *inlr = *on
С
                 return
C
                 endif
С
C* Write 'Hello World' to the file:
                 eval data = 'Hello World!'
С
                 callp
                         write(fd: %addr(data): %size(data))
С
C* Close the file:
  callp close(fd)
С
                 eval *inlr = *on
С
```

You typed all of that already? Wow, that was quick, good job! The instructions at the top tell you how to compile it. But, of course, you'll need to substitute the source library that you used for the "XXX" that I put in the comments.

For example, if your source library is called "TASTYSRC", you would type: **CRTBNDRPG CH1HELLO SRCFILE (TASTYSRC/QRPGLESRC) DBGVIEW (*LIST)**

Note: I don't recommend that name, it makes you want to stop and eat your computer.

Once you've succeeded in compiling the program, you can run it just by typing: CALL CH1HELLO

If a problem occurs when you run this program, you'll get a DSPLY message on the screen telling you that something failed. If nothing happens on your screen, rejoice! That's what we wanted to happen! In the next topic, we'll talk about what that program did, and verify that things worked.

1.6. Looking at our example from OS/400

In our hello world example, we defined a bunch of prototypes in our D-specs that told the system what APIs we wanted to call, and how the parameters were to be passed to them. If you don't understand how prototypes work, this might be a good time to whip out a book on prototypes and brush up.

```
fd = open('/helloworld': 1)
С
                      eval
                                           O CREAT+O TRUNC+O WRONLY:2
С
                                           (6 * 64) + (6 * 8) + (4) ) €
С
                                 fd < 0 4
                      if
С
                                 Msg = 'open failed!'
С
                      eval
С
                      dsplv
                                                msq
С
                      eval
                                 *inlr = *on
С
                      return
С
                      endif
```

When we get to the C-specs, you'll see that we are calling the open() API, like this:

- Here we tell the API that the file we want to open is in the root directory ("/") and is called "helloworld", and that the result of the call should be stored in the variable called "fd" (which is shorthand for "file descriptor")
- These flags tell the API that we want to create the file if it doesn't already exist (O_CREAT), truncate the file to zero bytes if it does exist (O_TRUNC) and that we're opening the only for writing to it (O_WRONLY) and not for reading.

"Truncating" the file can be thought of as "clearing" the file. All it does is delete any data that currently exists in the file.

• This confusing looking equation is the file's "mode", also called the "permissions" that we're giving to it. This tells the operating system who is allowed to read this file, who is allowed to write to it, and who is allowed to execute/search it (if it is a program or directory, respectively)

Because this "mode" is confusing, and hard-to-read, I use named constants to make it clearer. If you look at the previous example, you'll see that I used named constants to make it easier to read. I'll use some more standardized (though, arguably harder to read) constants in a later chapter.

We'll talk more about file modes in the upcoming chapter. For now, just know that we're giving read & write permission to ourselves, read & write permissions to anyone in our "group", and read-only permissions to everyone else.

• If an error occurred, the system will return a -1 to our "fd" variable. Therefore, we check for an error, and if one occurred, we'll report it to the user, and end the program.

We then proceed to write the string 'Hello World!' to the file, close it, and end the program.

To see if it worked, return to your trusty OS/400 command line and type: WRKLNK '/*'

OS/400 will bring up the "Work with Object Links" screen which shows you a list of all of the objects in the root directory of the IFS. (Depending on what's stored here on your system, there could be many pages of information.) Find the file called 'helloworld' in the list of object links. You'll note that it has a 'Type' of 'STMF', which means that it is a stream file.

Place a 5 next to it, and OS/400 will show you the contents of the stream file. You should now see that it contains the words "Hello World!" just as we expected. Press the F3 key to get back to the "Work With Object Links" screen.

Unless you have some nifty use for the helloworld file you created, let's delete it now, by putting a **4** next to it to delete it.

Yes, it's really that simple! Aside from the prototypes and maybe the file mode, this program is really short and easy to follow. Using this same technique, you can write almost anything to a stream file! We'll cover this more in depth, and show you how to read a stream file in the next chapter.

Chapter 2. The Basics of Stream Files

2.1. Opening Files with the open() API

2.1.1. Creating a header member

More than half of the code in our "Hello World" program was prototypes and other definitions. If we were C programmers, we wouldn't have to worry about this part of the coding, because IBM ships "header members" with OS/400 and ILE C/400 that C programmers can simply "include" at the top of their programs.

Since IBM doesn't do that for us as RPG programmers, what we'll do is create our own header member. We'll call it "IFSIO_H" which stands for "Integrated File System I/O Header Member," in that member, we'll put our prototypes, constants, etc. After we've done that, we'll no longer need to type those prototypes into every program. One /copy and we're done!

The first prototype that we'll add will be for the open() API.

2.1.2. Prototyping the open() API

The UNIX-type APIs manual shows us a prototype (written in C) for the open() API. It looks like this:

```
int① open②(const char *path③, int oflag④, ...⑤);
```

- The "int" here signifies what type of value the procedure returns. The "int" data type in C is identical to a "10I 0" variable in RPG IV.
- The word "open" here signifies the name of the sub-procedure that's being called. Like all things in C, procedure names are case-sensitive. This is important to us because, in RPG they are not case-sensitive! In fact, if we don't do anything special, the RPG compiler will convert the procedure name to all uppercase before binding. Therefore, when we make our RPG prototype, we'll use the EXTPROC() keyword to refer to open as an all-lowercase procedure name.
- This is where we specify the name of a path that we want to access. The "char" means character, and the "*" means pointer. So, what this procedure needs from us is a pointer that points to a character variable. You see, in C, character strings are implemented by specifying a starting point in memory, and then reading forward in memory until a x'00' (called a "null") is encountered.

Fortunately, RPG's options(*string) keyword, and the %str() built-in-function, make it easy for us to convert to and from C's string format.

- This is an integer ("int") which defines the flags that are used by the open() API. As I mentioned, the C "int" data type is equivalent to the "10I 0" data type in RPG.
- These three periods signify that any number of optional parameters may follow. This is difficult for us, since RPG prototypes need to know in advance what the data types of the parameters are.

Fortunately, if you read IBM's documentation, you find out that there are only two optional parameters that follow, and they are both unsigned integers. One is for specifying the mode, which is used when O_CREAT is passed as a flag, the other is for specifying a code page, which we will talk more about in chapter 5.

Now that we know how the C prototype works, we can write our own RPG prototype. Here's what I came up with:

| D | open | PR | 101 | 0 | extproc('open') |
|---|----------|----|-----|---|------------------------|
| D | path | | * | | value options(*string) |
| D | oflag | | 101 | 0 | value |
| D | mode | | 10U | 0 | value options(*nopass) |
| D | codepage | | 10U | 0 | value options(*nopass) |

Please add that prototype to your IFSIO_H member now, unless of course, you've decided to just install mine, in which case you already have it.

The first thing that you might notice is that all of the parameters are passed by value. That's because C is expecting to receive a pointer, an integer, an unsigned integer and another unsigned integer. If we passed these parameters by reference, it would actually receive 4 memory addresses, rather than receiving the actual data.

As a general rule, when a C program expects to receive something by value, it will simply list the data type followed by the variable. If it wants to receive the address of the variable, it will ask for a pointer to that variable by putting a "*" in front of it. For example "int oflag" is a request for an integer, passed by value, whereas "int *oflag" would be the same integer passed by reference.

Wait a minute! Wouldn't that mean that the "char *path" should be passed by reference, instead of by value?! Yes, that's true. In fact, we could've coded path as:

D path 1024A

However, if we did that, we'd have to assign a length to "path", and the C version allows path to be of any length. The trick is, passing a pointer by value is the same thing as passing a variable by reference. In either case, what actually gets passed from procedure to procedure is an address in memory. But, if we use a pointer by value, and we use "options(*string)", the RPG compiler will automatically allow any length string, and will automatically convert it to C's format by adding the terminating "null" character at the end. This saves us some work.

Finally, you'll notice that the mode and codepage parameters are set up with "options(*nopass)". This means that they're optional, and we only need to pass them when we need them.

2.1.3. The path parameter

path is pretty self-explanatory. It is the path to the file in the IFS that we want to open. In the IFS, the "/" character is used to separate the different components of the path. The first component may be a "/" to signify the root directory. Thereafter, each section of the path refers to a directory name until we reach the last component, which specifies the filename.

For example, consider the following path name:

/ifsebook/chapter2/examples/myfile.txt

The leading "/" means to start at the root of the IFS. If it was not specified, the path name would actually start at whatever our current working directory was, and continue from there. But since it has a "/" we're telling it that the path to the file actually starts at the root of the system.

The word "ifsebook" refers to a directory. The word "chapter2" refers to a sub-directory that's inside the "ifsebook" directory. The word "examples" refers to another sub-directory, this one is inside the "chapter2" directory, and finally, "myfile.txt" refers to the object that's in the "examples" directory.

Let's try another, somewhat more familiar, example:

/QSYS.LIB/qgpl.lib/qrpglesrc.file/proof.mbr

This tells us to go back to the "/" root directory, then look at the QSYS.LIB directory (which is also known as the QSYS library) and that within that directory is a directory called the qgpl.lib directory (which is also known as the QGPL library) and within that, there's a file called QRPGLESRC which contains a member called "PROOF".

2.1.4. The oflag parameter

oflag is where we specify the options we want to use when opening the file. What we're actually passing here is a string of 32 bits, each of which specifies a different option. The rightmost bit specifies "Read only", then moving one bit to the left, that bit specifies "Write only", and the next bit specifies "reading and writing" and the next bit specifies "create the file if it doesn't exist," etc.

In order to make our lives easier, rather than actually specifying the bits manually, we define a series of flags, that when added together, will turn on the bits that we desire.

For example, we use the number 8 to signify "create if the file doesn't exist" and the number 2 to signify "write only". This makes more sense if you convert those numbers to binary. The decimal number 8 is 1000 in binary. The decimal number 2 is 10 in binary. So you see, when we specify the number 8, we actually specify that we want the 4th bit (counting from the right) to be turned on. When we specify 2, we are specifying that the 2nd bit be turned on. If we add those two numbers together, 8+2=10. If you convert the decimal 10 to binary you get 1010 (bits 4 and 2 are both on). Because each of these numbers turns on a single bit, we refer to them as "flags", and they supply us with a convenient way to which options we want to pass to the open() API.

So that we don't need to mess with the bit values later in this book, let's add those flags to our IFSIO_H member now. This is what we need to add:

```
D* Flags for use in open()
D*
D* More than one can be used -- add them together.
D*
                            Reading Only
D O_RDONLY
           С
                       1
                            Writing Only
D*
D O_WRONLY
           С
                       2
                            Reading & Writing
D*
D O RDWR
           С
                       4
D*
                            Create File if not exist
D O_CREAT
           С
                       8
                            Exclusively create
D*
```

| D O_EXCL | С | 16 |
|--------------|---|---------------------------|
| D* | | Truncate File to 0 bytes |
| D O_TRUNC | С | 64 |
| D* | | Append to File |
| D O_APPEND | С | 256 |
| D* | | Convert text by code-page |
| D O_CODEPAGE | С | 8388608 |
| D* | | Open in text-mode |
| D O_TEXTDATA | С | 16777216 |

2.1.5. The mode parameter

mode is used to specify the access rights that this file will give to users who want to work with it. Like the "oflag" parameter, this parameter is treated as a series of bits. The rightmost 9 bits are the ones that we're concerned with, and they're laid out like this:

| user: | owner | group | other |
|---------|-------|-------|-------|
| access: | RWX | RWX | R W X |
| bit: | 987 | 654 | 321 |

These bits specify Read, Write and Execute access to 3 types of users. The first is the file's owner, the second is users with the same group profile as the file's owner, and the third is all other users.

For example, if I wanted to specify that the owner of the file can read and write to the file, that people in his group can only read the file, and that everyone else has no access at all, I'd specify the following bits: 110 100 000. If you look at those bits as the binary number 110100000, and convert it to decimal, you'd get 416. So, to assign those permissions to the file, you'd call open() with a 3rd parameter of 416.

Just as we did for the "oflags" parameter, we'll also specify bit-flags for the mode, which we can add together to make our programs easier to read. Please add these to your IFSIO_H member:

```
D*
    Mode Flags.
D*
       basically, the mode parm of open(), creat(), chmod(),etc
D*
       uses 9 least significant bits to determine the
D*
       file's mode. (peoples access rights to the file)
D*
D*
        user:
                owner group other
        access:
                RWX RWX
                            RWX
D*
        bit:
                 876
                       543
                              2 1 0
D*
D*
D* (This is accomplished by adding the flags below to get the mode)
D*
                              owner authority
D S_IRUSR
            С
                          256
D S_IWUSR
            С
                          128
D S_IXUSR
           С
                          64
D S_IRWXU
           С
                          448
D*
                              group authority
```

| D S_IRGRP | С | 32 | |
|-----------|---|----|--------------|
| D S_IWGRP | С | 16 | |
| D S_IXGRP | С | 8 | |
| D S_IRWXG | С | 56 | |
| D* | | | other people |
| D S_IROTH | С | 4 | |
| D S_IWOTH | С | 2 | |
| D S_IXOTH | С | 1 | |
| D S_IRWXO | С | 7 | |

Now, instead of specifying "416", we can simply add together S_IRUSR+S_IWUSR+S_IRGRP, which specifies "read access for user", "write access for user" and "read access for group", respectively.

2.1.6. The codepage parameter

If you specify the O_CODEPAGE flag in the oflag parameter, you must use this parameter to specify which code page the file will be assigned.

We will talk about that more in Chapter 5, in our discussion of text files.

2.1.7. The return value of the open() API

The *return value* of the open() API is a "file descriptor". It is an integer that we will pass to all of the other IFS APIs that we call so that they know which file we are referring to. If something goes wrong, and the system is not able to open the file that we requested, it will return a value of -1 instead of a file descriptor. So, whenever we call open() we will check for this, and treat -1 as an error.

2.1.8. Code snippet showing the use of the open() API

Here's a code snippet that uses the open() API:

| c c | eval | <pre>path = '/QIBM/UserData/OS400/DirSrv' + '/slapd.conf'</pre> |
|--------|-------|---|
| с | eval | <pre>flags = O_WRONLY + O_CREAT + O_TRUNC</pre> |
| С | eval | <pre>mode = S_IRUSR + S_IWUSR</pre> |
| С | | + S_IRGRP |
| С | eval | <pre>fd = open(%trimr(path): flags: mode)</pre> |
| С | if | fd < 0 |
| С | goto | bomb_out |
| С | endif | |

2.2. Closing a file with the close() API

If your head is spinning after reading about the open() API, you'll be glad to know that the close() API is comparatively simple.

The close() API is used to close a file that we opened when we called the open() API. Here is the C language prototype for close(), as it is listed in IBM's UNIX-type APIs manual:

int close(int fildes);

Simple, yes? The "int" is the return value. The "close" is the name of the procedure. The "int fildes" is the only parameter to the procedure, and it's just an integer, which is "10I 0" in RPG.

So, the RPG prototype will look like this:

| D | close | PR | 101 | 0 | <pre>extproc('close')</pre> |
|---|--------|----|-----|---|-----------------------------|
| D | fildes | | 101 | 0 | value |

See? It returns a "10I 0", because the C prototype returned an "int". It accepts one parameter, which is also a "10I 0" because the C prototype's only parameter was an "int". The parameter is passed by value because we don't want to pass the address, and we use "extproc()" to make sure the compiler doesn't try to call "CLOSE" instead of "close".

Great.

There's one small problem. This same close() API is used by both socket connections (TCP/IP communications API) and also by the IFS APIs that we're using. That's a problem because if you ever tried to use both sockets and IFS in the same program, the definitions would conflict, and the program wouldn't compile.

So, we're going to use a little "compiler directive" magic to make sure that the two close() prototypes never conflict, by making the prototype look like this:

```
D/if not defined(CLOSE_PROTOTYPE)
D close PR 10I 0 extproc('close')
D fildes 10I 0 value
D/define CLOSE_PROTOTYPE
D/endif
```

And then, when the day comes that we make a header member for sockets, we'll have to remember to also put that same /define logic in the sockets header member. Do you see how it works? Pretty cool eh?

Here's an example of calling the close() API:

```
c callp close(fd)
```

2.3. Writing streams with the write() API

The write() API is used to write bytes to a stream file.

The reason that I say "bytes" as opposed to saying "letters" or "words" is that a byte containing any value can be written. We are not limited to just alphanumeric strings of text. We can write the contents of a packed decimal variable, for example, or an entire data structure.

All you have to tell write() is an area of memory (and again, it doesn't care what's in that area) and length. It copies the bytes from memory to disk.

Here's what the C language prototype of write() looks like, as printed in the UNIX-type APIs manual:

int① write(int fildes②, const void *buf③, size_t nbyte④);

Now that's a sexy looking API!

- The return value is an "int", which is a 32-bit signed integer, identical to the RPG "10I 0" data type. The write() API returns the number of bytes written to disk. If something goes wrong, this number will be smaller than the amount we told it to write, so we can use this to detect errors, as well.
- The first parameter is also an integer, and this one represents the file descriptor, which is the value that we got from the open() API.
- Something new! What the heck could "const void *buf" mean? Well, my friend, it's quite simple. The "const" is just like the RPG "const" keyword. It simply means that the API cannot and will not change the contents of the parameter. But does "void *" mean that it's a pointer to a void? No, not really. It means that this pointer can point to anything. It doesn't have to be a character, a number, or a structure. It can point to any byte value in memory.
- And finally, we have "size_t nbyte". It's pretty clear that "nbyte" is short for "number of bytes to write". But what is that funny "size_t" thing?

It's a "user-defined" type. It's similar in some ways to the "like" keyword in RPG. The idea is that on different platforms, the way that a byte size is stored may be different. To write code that's reusable, they have a header member called "sys/types.h" which describes the actual data types of things like "size_t", and then when you change to a different platform and use different header files, the program compiles and works on that platform as well.

On the AS/400, size_t is defined to be a 32-bit unsigned integer. In other words, it's the same as the RPG "10U 0" data type.

Now that we've covered that, here's the RPG version of the prototype. You'll want to add this to the IFSIO_H header member:

| Dw | rite | PR | 101 | 0 | <pre>extproc('write')</pre> |
|----|--------|----|-----|---|-----------------------------|
| D | fildes | | 101 | 0 | value |
| D | buf | | * | | value |
| D | nbyte | | 10U | 0 | value |

See? Not so bad. Just a couple of integers and a pointer. no sweat.

Here's a code snippet showing a program that calls the write() API:

| С | eval | wrdata = 'THE QUICK BROWN FOX JUMP' |
|---|-------|--|
| С | if | <pre>write(fd: %addr(wrdata): %size(wrdata))</pre> |
| С | | < %size(wrdata) |
| С | goto | |
| С | endif | |

2.4. Reading a stream file with the read() API

The read() API is the exact opposite of the write() API. It reads bytes of data from a stream file, and stores them into the area of memory that you point it to.

Here's the C language prototype of read():

```
int read(int fildes, void *buf, size_t nbyte);
```

This prototype is so much like the write() API that I won't even describe the process of converting it from C to RPG. From reading about write(), you should already understand. So, without any more long-winded ramblings, here's the RPG prototype:

| D | read | PR | 10I (|) extproc('read') |
|---|--------|----|-------|-------------------|
| D | fildes | | 10I (|) value |
| D | buf | | * | value |
| D | nbyte | | 10U (|) value |

Make sure you put that in your copy of the IFSIO_H member.

In many ways, read() is very similar to write(). Like it's counterpart, it can be used to work with any byte values. It does not care if the data is numeric or character or part of a graphic or sound format. And, like write() it takes 3 parameters, and returns an integer. There are, however, some crucial differences:

- Obviously, it reads instead of writes. Otherwise it would be silly to call it "read!"
- Note that the "buf" argument is no longer marked as "const". That means that the API definitely can change the contents of the variable that buf points to!

In fact, that's where read will store the information that it loads from the stream file.

• The "nbyte" parameter tells read the size of the variable that the "buf" parameter is pointing to. It's true that read() will try to read that many bytes from the disk, but if you're at the end of the stream file, read() may read fewer bytes than you've specified in the "nbyte" argument. Don't treat that as an error!

You call the read() API like this:

| С | eval | <pre>len = read(fd: ptr2buf: %size(buf))</pre> |
|---|-------|--|
| С | if | len < 1 |
| с | goto | no_more_to_read |
| С | endif | |
| | | |

2.5. Example of writing and reading data to a stream file

The last few topics have probably given you lots of new things to think about. It's time to play with them!

First of all, let's create a directory to write all of our stream files to. This will keep us from cluttering up the root directory of the IFS with our tests. Let's call this new directory "ifstest." We can create it by typing the following command at our OS/400 command-line:

CRTDIR DIR('/ifstest')

If you get an error that you do not have sufficient authority to create it, or something like that, you may need to speak with your system administrator. Tell him that you need a sandbox to play in!

Here's a program which both writes and reads data from a stream file. It also demonstrates one of the properties of a stream file -- the data is stored as a continuous stream of bytes, not in records.

Take a look at it, guess what you think it does, then try it out. It's pretty cool!

| * CH2WRRD: Exampl * (From Chap 2) * * To compile: | e of writir | ng & reading data to a stream file |
|--|---------------|---|
| * CRTBNDRPG CH2 | WRRD SRCFII | LE(xxx/QRPGLESRC) DBGVIEW(*LIST) |
| * | | |
| H DFTACTGRP(*NO) A | CTGRP (*NEW) | |
| D/copy IFSEBOOK/QR | PGLESRC, IFS | SIO_H |
| D fd | S | 10I 0 |
| D wrdata | S | 24A |
| D rddata | S | 48A |
| D flags | S | 100 0 |
| D mode | S | 10U 0 |
| D Msg | S | 50A |
| D Len | S | 101 0 |
| C* Example of writ | ing data to | ************************************** |
| C | CVUI | |
| С | eval | <pre>mode = S_IRUSR + S_IWUSR</pre> |
| С | | + S_IRGRP |
| С | | + S_IROTH |
| | | |
| С | eval | <pre>fd = open('/ifstest/ch2_test.dat':</pre> |
| С | | <pre>flags: mode)</pre> |
| С | if | fd < 0 |
| С | eval | <pre>Msg = 'open(): failed for writing'</pre> |
| С | dsply eval | Msg *inlr = *on |
| С | return | * Inlr = * On |
| c c | endif | |
| C | enarr | |
| C* Write some data | | |
| C | eval | wrdata = 'THE QUICK BROWN FOX JUMP' |
| C | callp | |
| | - | |
| C* Write some more | data | |

```
eval
                         wrdata = 'ED OVER THE LAZY GIRAFFE'
С
С
                callp
                         write(fd: %addr(wrdata): %size(wrdata))
C* close the file
                callp
                         close(fd)
С
C* Example of reading data from a stream file
eval
                        flags = O_RDONLY
С
                        fd = open('/ifstest/ch2_test.dat':
                eval
С
С
                                 flags)
                if
                         fd < 0
С
                        Msg = 'open(): failed for reading'
С
                eval
                dsply
                                    Msq
С
                eval
С
                         *inlr = *on
С
                return
                endif
С
                eval
                        len = read(fd: %addr(rddata):
С
                                     %size(rddata))
С
                        Msg = 'Length read = ' +
С
                eval
                             %trim(%editc(len:'M'))
С
С
     Msq
                dsply
                                    rddata
С
                dsply
С
                callp
                        close(fd)
                         *inlr = *on
                eval
С
                return
С
```

As before, there are instructions on compiling the program near the top of the source. Give it a try. I'll be right back, I'm going to go get a cold beverage.

Ahhhh... lime soda.

2.6. Error handling

The world is an imperfect place. Things go wrong. Sometimes a file can't be opened, or sometimes a tyrannical system administrator won't let us access something. It can be rough.

One of the problems with the example programs that we've written so far is that, although they detect when something went wrong, they couldn't tell us what the problem was. They know something happened, but they don't know what.

2.6.1. Retrieving the error number.

Like most of the UNIX-type APIs, our IFS functions return their error information using the C language "errno" variable. The idea is that there is a global variable called "errno" which a C program can check after something has gone wrong. The result is an integer that corresponds to a specific error message.

On the AS/400, the "errno" variable is actually returned by a sub-procedure that, for C programmers, gets called behind-the-scenes. So, for us to check errno, all we have to do is call that sub-procedure, and get the return value.

The sub-procedure that returns error information is called "__errno" and is part of the ILE C runtime library which is installed on every AS/400. The C language prototype for "__errno" looks like this:

```
int *__errno(void);
```

What that means is that the procedure is called __errno, and it returns a ("int *") pointer to an integer. The "void" signifies that there are no parameters.

In RPG, you can't start a sub-procedure name with the underscore character, so we'll add another symbol to the front of the prototype to make it work. The result looks like this:

D@_errno PR * ExtProc('_errno')

Now, you'll note that although we're looking for an integer, this procedure actually returns a pointer. Yech! So what we'll do is create a simple sub-procedure that gets an integer from the area of memory that the pointer points at. That's a very simple sub-procedure, and it looks like this:

```
P errno
                  В
                                 10I 0
D errno
                  ΡТ
D p_errno
                  S
                                   *
D retval
                  S
                                 10I 0 based(p_errno)
С
                    eval
                               p errno = @ errno
С
                    return
                               retval
Ρ
                   Е
```

2.6.2. What does the error number mean?

So, now we know that errno can be called, and it will give us an integer that tells us which error has occurred. But, what does the number mean? For example, if we got back the number 3401, how would we know what went wrong?

In C, there's a source member which programmers use that contains constants for each error number. For example, it will define the constant EACCES to be the number 3401. The C program can compare errno to EACCES, and if they match, it knows that the user does not have enough access (or "authority") to carry out the function.

In fact, if you look at the text in the IBM Information Center that explains (for example) the write() API, you'll see that under "Error Conditions" it says "If write() is not successful, errno usually indicates one of the following errors . . ." and then goes on to list errors like [EACCES] and [ENOSPC]. These error conditions are nothing more than the named constants that I mentioned above.

Since the "errno" stuff can be used by other APIs besides the ones that this book covers, we will place these in their own header member. That way, you can include them into your future programs without also including the code that's IFS-specific.

I've called my /copy member "ERRNO_H". If at all possible you should consider using the one that I provide with this book. In it, I've put RPG named constants corresponding to all the values of errno that I know about. Since it would be tedious for you to find all of these values and type them in, you may as well just use mine!

2.6.3. Getting a human-readable error message

In addition to the named constants for each error number, it's useful to have a "human-readable" error message that corresponds to each error number. For example, when you want to print a message on the screen explaining what went wrong, you'd probably rather say "No such path or directory" rather than "Error 3025 has occurred."

The ILE C/400 runtime library contains a procedure called "strerror()" for this purpose. When you call strerror() with an error number as a parameter, it returns a pointer to a variable length, null-terminated, error message. Here's the C and RPG prototype for strerror():

```
char *strerror(int errnum);
D strerror PR * ExtProc('strerror')
D errnum 10I 0 value
```

In addition to strerror(), you can also view each error number as a message in an OS/400 message file. The QCPFMSG message file contains all of the C error numbers prefixed by a "CPE". For example, the error ENOENT is 3025. If you type **DSPMSGD CPE3025 MSGF (QCPFMSG)** at the command prompt, it will show you a message that says "No such path or directory." Likewise, if you looked up CPE3401, you'd see the human-readable message "Permission denied."

2.6.4. Utilities for communicating errors

As you may already know, OS/400 programs usually return error information from one program to another by sending a "program message". When an error occurs which causes a program to fail, the program usually sends back a program message of type "escape" to it's caller.

For the sake of making error handling easier, we will create two simple sub-procedures that we can use to send back "escape messages", and add these to our ERRNO_H file, so all of our programs can use them.

The first utility is called "die". It will send back any user supplied error message under a message number of CPF9897. This is useful for supplying simple text error messages in our example programs. Here's the code:

| Рc | die | В | | |
|----------|-----------------------|----|-----------|-----------------------------|
| Dc | die | PI | 1N | |
| D | msg | | 256A | const |
| | | | | |
| | | | | |
| Dζ | OMHSNDPM | PR | | ExtPgm('QMHSNDPM') |
| D Ç D | QMHSNDPM MessageID | PR | 7A | ExtPgm('QMHSNDPM') Const |
| - | ~ | PR | 7A 20A | 5 |

| D MsgDtaLen D MsgType D CallStkEnt D CallStkCnt D MessageKey D ErrorCode | | 10I 0 Const 10A Const 10A Const 10I 0 Const 4A 256A |
|---|---------------------------------|--|
| D dsEC D dsECBytesP D dsECBytesA D dsECMsgID D dsECReserv D dsECMsgDta | DS 1 5 9 16 17 | 4I 0 inz(%size(dsEC)) 8I 0 inz(0) 15 16 256 |
| D MsgLen D TheKey | S S | 10I 0 4A |
| с / / с с | checkr if return endif | msg MsgLen MsgLen<1 *off |
| с с с | callp | QMHSNDPM('CPF9897': 'QCPFMSG *LIBL': Msg: MsgLen: '*ESCAPE': '*': 3: TheKey: dsEC) |
| C P | return E | *off |

The other utility function is called "EscErrno". We will pass an error number as an argument to this function, and it will send back the appropriate CPExxxx error message as an escape message to the calling program.

EscErrno is useful when we want our programs to crash and report errors that the calling program can monitor for individually. For example, a calling program could be checking for CPE3025, and handle it separately than CPE3401.

Here is the code for EscErrno:

| Ρ | EscErrno | В | | | |
|---|------------|----|------|---|--------------------|
| D | EscErrno | PI | 1N | | |
| D | errnum | | 10i | 0 | value |
| | | | | | |
| D | QMHSNDPM | PR | | | ExtPgm('QMHSNDPM') |
| D | MessageID | | 7A | | Const |
| D | QualMsgF | | 20A | | Const |
| D | MsgData | | 1A | | Const |
| D | MsgDtaLen | | 101 | 0 | Const |
| D | MsgType | | 10A | | Const |
| D | CallStkEnt | | 10A | | Const |
| D | CallStkCnt | | 101 | 0 | Const |
| D | MessageKey | | 4A | | |
| D | ErrorCode | | 256A | | |
| | | | | | |
| D | dsEC | DS | | | |

| D dsECBytesP | 1 | 4I 0 inz(%size(dsEC)) |
|--------------|--------|----------------------------------|
| D dsECBytesA | 5 | 8I 0 inz(0) |
| D dsECMsgID | 9 | 15 |
| D dsECReserv | 16 | 16 |
| D dsECMsgDta | 17 | 256 |
| D TheKey | S | 4A |
| - | | |
| D MsgID | S | 7A |
| | | |
| С | move | errnum MsgID |
| C | movel | 'CPE' MsgID |
| | | |
| С | callp | QMHSNDPM(MsgID: 'QCPFMSG *LIBL': |
| С | | ' ': 0: '*ESCAPE': |
| С | | '*': 3: TheKey: dsEC) |
| | | |
| С | return | *off |
| P | Е | |

What I've done in my ERRNO_H is put the codes for the all of the procedures (errno, die, and escerrno) at the bottom, and enclosed them in these:

```
/if defined(ERRNO_LOAD_PROCEDURE)
.... procedure code goes here ....
/endif
```

This allows us to include all of the error handling code in our programs by copying the header member twice, once without the "errno_load_procedure" symbol defined, which goes in our D-specs, and once with the "errno_load_procedure" symbol defined, which goes where our sub-procedures go.

2.7. Our last example with error handling added

Here's an example of the error handling code that we discussed in the last section. All I did here is take the sample code that we wrote in section 2.5 above, and add error checking to it. When something goes wrong, the program calls die() to signal an abnormal end.

```
* CH2ERRNO: Example of writing & reading data to a stream file
* with error handling. (This is the same as CH2WRRD except
* for the error handling)
* (From Chap 2)
*
* To compile:
* CRTBNDRPG CH2ERRNO SRCFILE(xxx/QRPGLESRC) DEGVIEW(*LIST)
*
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSeBOOK/QRPGLESRC,IFSIO_H
D/copy IFSeBOOK/QRPGLESRC,ERRNO_H
```

| D fd S | 5 | 10I 0 |
|---------------------|-----------------------|---|
| D wrdata S | 5 | 24A |
| D rddata S | 5 | 48A |
| D flags | 5 | 100 0 |
| D mode | 5 | 100 0 |
| | 5 | 250A |
| - | 6 | 50A |
| 5 | | 10I 0 |
| D Len | 5 | 101 0 |
| | | |
| | | ************** |
| C* Example of writ: | ing data to |) a stream file |
| C****** | * * * * * * * * * * * | ******** |
| С | eval | <pre>flags = O_WRONLY + O_CREAT + O_TRUNC</pre> |
| | | - |
| С | eval | <pre>mode = S_IRUSR + S_IWUSR</pre> |
| | evai | + S_IRGRP |
| С | | _ |
| С | | + S_IROTH |
| | | |
| С | eval | <pre>fd = open('/ifstest/ch2_test.dat':</pre> |
| С | | flags: mode) |
| С | if | fd < 0 |
| С | eval | ErrMsq = %str(strerror(errno)) |
| C | callp | die('open() for output: ' + ErrMsg) |
| | endif | die (open () for output. • Hilling) |
| С | enarr | |
| ~ ~ ~ | | |
| C* Write some data | | |
| С | eval | wrdata = 'THE QUICK BROWN FOX JUMP' |
| С | if | <pre>write(fd: %addr(wrdata): %size(wrdata))<1</pre> |
| С | eval | ErrMsg = %str(strerror(errno)) |
| С | callp | close(fd) |
| С | callp | <pre>die('open(): ' + ErrMsg)</pre> |
| C | endif | |
| 6 | CHAIL | |
| | | |
| C* Write some more | | |
| С | eval | wrdata = 'ED OVER THE LAZY GIRAFFE' |
| С | if | <pre>write(fd: %addr(wrdata): %size(wrdata))<1</pre> |
| С | eval | ErrMsg = %str(strerror(errno)) |
| С | callp | close(fd) |
| С | callp | die('open(): ' + ErrMsg) |
| С | endif | |
| - | | |
| C* close the file | | |
| | | |
| С | callp | close(fd) |
| | | |
| C************ | * * * * * * * * * * * | * |
| C* Example of read: | ing data fr | com a stream file |
| C***** | * * * * * * * * * * * | ****** |
| С | eval | flags = O_RDONLY |
| | | - — |
| | | |
| C | ا دىرم | $fd = open(1/ifstest/ch^2 + ost dot')$ |
| с | eval | <pre>fd = open('/ifstest/ch2_test.dat':</pre> |
| с с с | eval if | <pre>fd = open('/ifstest/ch2_test.dat':</pre> |

```
ErrMsg = %str(strerror(errno))
С
                     eval
С
                     callp
                                die('open() for input: ' + ErrMsg)
                     endif
С
                                len = read(fd: %addr(rddata):
                     eval
С
                                               %size(rddata))
С
                     if
                                len < 1
C
                                ErrMsg = %str(strerror(errno))
С
                     eval
С
                     callp
                                close(fd)
                     callp
                                die('read(): ' + ErrMsg)
С
                     endif
С
                               Msg = 'Length read = ' +
С
                     eval
                                       %trim(%editc(len:'M'))
С
      Msg
                     dsply
С
                     dsply
                                               rddata
С
С
                     callp
                                close(fd)
                     eval
                                *inlr = *on
С
С
                     return
 /DEFINE ERRNO_LOAD_PROCEDURE
 /COPY IFSEBOOK/QRPGLESRC, ERRNO_H
```

The only reason that I used die() instead of EscErrno() to handle errors in this program is that with die() I can easily add text explaining where the error occurred. In the next section, I'll give an example of EscErrno().

2.8. Example of writing raw data to a stream file

As I've mentioned in previous sections, stream files can be used for any byte values, not just for words and other text. As a proof of concept, I thought it might be fun to generate a very small MS-DOS program as a stream file.

Programs under OS/400 are stored in *PGM objects. You can't directly open and manipulate a *PGM object on the AS/400. You have to write source code, and let the compiler compile it.

However, on the PC, all objects are stored as stream files. It doesn't matter if it's a data file, a program, an audio file, etc. Every object is stored as a stream file!

To prove that, here's an RPG program that actually generates a PC program. The stream file that it outputs can be downloaded to your PC, and run. (Provided, of course, that the PC is able to run MS-DOS programs).

```
* CH2RAWDTA: Example of writing non-text to a stream file
* (From Chap 2)
*
* To compile:
* CRTBNDRPG CH2RAWDTA SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
*
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSEBOOK/QRPGLESRC,IFSIO_H
D/copy IFSEBOOK/QRPGLESRC,ERRNO_H
```

| D fd | S | 101 0 |
|----------|--------|--|
| D wrdata | S | 79A |
| D err | S | 10I 0 |
| | | |
| С | eval | wrdata = $x'B409BA0C01CD21B8004CCD21' +$ |
| С | | x'416C6C206F626A6563747320' + |
| С | | x'6F6E20746865205043206172' + |
| С | | x'652073746F72656420696E20' + |
| С | | x'2273747265616D2066696C65' + |
| С | | x'73222C206576656E2070726F' + |
| С | | x'6772616D732124' |
| | | |
| С | eval | <pre>fd = open('/ifstest/littlepgm.com':</pre> |
| С | | O_WRONLY+O_CREAT+O_TRUNC: |
| С | | S_IRUSR + S_IWUSR + S_IXUSR |
| С | | + S_IRGRP + S_IXGRP |
| С | | + S_IROTH + S_IXOTH) |
| С | if | fd < 0 |
| С | callp | EscErrno(errno) |
| С | endif | |
| | | |
| С | if | <pre>write(fd: %addr(wrdata): %size(wrdata))</pre> |
| С | | < %size(wrdata) |
| С | eval | err = errno |
| С | - | close(fd) |
| С | - | EscErrno(err) |
| С | endif | |
| | | |
| С | callp | close(fd) |
| | 2 | |
| С | | <pre>*inlr = *on</pre> |
| С | return | |
| | | |

/DEFINE ERRNO_LOAD_PROCEDURE /COPY IFSEBOOK/QRPGLESRC,ERRNO_H

This program outputs a stream file called littlepgm.com into our /ifstest directory. To run this program you'll need to transfer it to your PC. One way to do this, if your system is set up for it, would be to map a network drive to your AS/400. Another good choice would be to FTP the file to your PC. If you decide to use FTP, make sure that you use binary mode.

Once you've got it on your PC, you should run the program from an MS-DOS prompt.

Chapter 3. Other simple, but helpful IFS commands

3.1. Checking existence and permissions to files

In the last chapter, we covered the basics of how stream files work. It's all down hill from here!

One question that I've been asked many times is: "I use the CHKOBJ command in my CL programs. How can I do the same thing with a file in the IFS?"

The answer is the access() API. Access() can be used to check two things: whether the file exists, and whether it's accessible for reading, writing or execution.

The C-language prototype for the access() API looks like this:

int access(const char *path, int amode);

The prototype is quite simple, and I think by now you're already getting the hang of it, so without further ado, here's the RPG version:

| 0 | Dа | ccess | PR | 10I | 0 | ExtProc('access') |
|---|----|-------|----|-----|---|------------------------|
| 0 | D | Path | | * | | Value Options(*string) |
| 0 | D | amode | | 10I | 0 | Value |

Please add this to the IFSIO_H copy member, if you're typing it in yourself.

• The access API returns an integer which can either be 0 if the file is accessible, or -1 if for some reason it is not. Like most UNIX-type APIs, we can check errno after calling access to find out why the file wasn't accessible.

• This is the path name of the IFS object that we want to check the accessibility of.

• This is the access that we want to check. This is another one of those "bit flag" fields, similar to the ones we've been using in the open() API.

The *amode* parameter uses the rightmost 3 bits of the parameter to determine which access we want to check. If the right-most bit is on, access() checks for execute authority. The next bit to the left checks for write access, and the 3rd bit from the right checks for read access.

If none of the bits in the amode parameter are set, the API will only check to see if the object exists.

Just like we did for the other bit-flags that we've used, we will define named constants to both to make our code easier to follow, and also to match the constants that are already defined for the C programmers.

| D F_OK | С | 0 |
|--------|---|---|
| D R_OK | С | 4 |
| D W_OK | С | 2 |
| D X OK | С | 1 |

Here's a sample of calling access() in an RPG program:

| С | if | access(%trimr(myfile): F_OK) < 0 |
|------------------|-----------------------------|--|
| С | eval | err = errno |
| С | if | err = ENOENT |
| С | callp | <pre>die('Errrm can"t find that file!')</pre> |
| С | else | |
| С | callp | die(%str(strerror(err))) |
| С | endif | |
| С | endif | |
| | | |
| | | |
| С | if | access(%trimr(myfile): R_OK) < 0 |
| c c | if eval | access(%trimr(myfile): R_OK) < 0 err = errno |
| | | - |
| С | eval | err = errno |
| C C | eval if | err = errno err = EACCES |
| с с с | eval if | err = errno err = EACCES die('It"s there, but YOU can"t read ' + |
| с с с | eval if callp | <pre>err = errno err = EACCES die('It"s there, but YOU can"t read ' + 'it! Nyaahh! Nyaahh!')</pre> |
| C C C C | eval if callp else | <pre>err = errno err = EACCES die('It"s there, but YOU can"t read ' + 'it! Nyaahh! Nyaahh!')</pre> |

3.2. Example of checking for an object in the IFS

Here's an example that's also a useful utility. We will create a command called "CHKIFSOBJ" which works like the OS/400 CHKOBJ command, except that it operates on an IFS object.

CHKOBJ allows you to check if the file exists, and also allows you to optionally check if you have authority to use it. So, our IFS version will do the same. To do that, we need two parameters, the path name of the object to check, and the authority to check for.

Here's the command definition to do that:

| CMD | PROMPT('Check for IFS Object') |
|------|--|
| PARM | KWD(OBJ) TYPE(*CHAR) LEN(640) MIN(1) + |
| | PROMPT('Object') |
| PARM | <pre>KWD(AUT) TYPE(*CHAR) LEN(10) RSTD(*YES) +</pre> |
| | DFT(*NONE) VALUES(*NONE *R *RW *RX *RWX + |
| | *W *WX *X) PROMPT('Authority') |

Take a look at the "AUT" parameter. It allows you to specify "*NONE" if you just want to see if an object exists, or to check for read, write, execute or any combination of them. It's just like access(), really!

Now, here's the RPG code that our command will run:

* CH3CHKOBJ: Example of checking for an object in the IFS

* (From Chap 3) * To compile: CRTBNDRPG CH3CHKOBJ SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST) CRTCMD CMD(CHKIFSOBJ) PGM(CH3CHKOBJ) SRCFILE(xxx/QCMDSRC) * H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('OC2LE') D/copy IFSEBOOK/QRPGLESRC, IFSIO_H D/copy IFSEBOOK/QRPGLESRC, ERRNO_H D Path 640A S 10A D Authority S D AMode S 10T 0 ** Warning: call this program from the command. If you call it directly, because "Path" is larger than 32 bytes. * * * * See http://faq.midrange.com/data/cache/70.html * * С *entry plist Path С parm Authority С parm C* First, just check if the file exists: if Access(%trimr(Path): F_OK) < 0 С С callp EscErrno(errno) С endif C* Next, check if the current user has authority: if Authority <> '*NONE' С eval amode = 0С if scan('R':Authority) > 0С С eval amode = amode + R_OK С endif if scan('W':Authority) > 0С eval amode = amode + W_OK С endif С if scan('X':Authority) > 0С eval amode = amode + X_OK С С endif С if access(%trimr(Path): amode) < 0</pre> С callp EscErrno(errno) endif С endif С *inlr = *on С eval

/DEFINE ERRNO_LOAD_PROCEDURE

/COPY IFSEBOOK/QRPGLESRC, ERRNO_H

3.3. Changing permissions on an existing IFS Object

The API that allows us to change the permissions of an IFS object is called "chmod", which stands for "change mode". Here's the C-language prototype for chmod(), along with my RPG equivalent:

| | int chr | nod(const | char | *path, | mode_t | mo | ode) | |
|---|---------|-----------|------|--------|--------|----|--------|------------------|
| | | | | | | | | |
| D | chmod | | PR | | 101 | 0 | ExtPro | oc('chmod') |
| D | path | l | | | * | | Value | options(*string) |
| D | mode | è | | | 10U | 0 | Value | |

The "mode" parameter works exactly the same as the "mode" parameter on the open() API. You use the same bit-flags to assign permissions to the "owner", the "group" and "others".

The difference between chmod() and open() is that chmod() does not open or create a new file, but instead changes the access permissions on an existing IFS object.

For example, let's say that you had already run the example program from chapter 2 called "CH2RAWDTA", and you know that it wrote an object to disk called "littlepgm.com". But now, you decided that you didn't want Bill from Accounting to be able to download your program! Since it's already there, you'd want to remove "read" permissions from the object.

To do that, you'd do something like this:

| С | if | <pre>chmod('/ifstest/littlepgm.com':</pre> |
|---|-------|--|
| С | | S_IRUSR + S_IWUSR + S_IXUSR) < 0 |
| С | callp | EscErrno(errno) |
| С | endif | |

We assigned Read, Write and Execute authority to the file's owner, but we gave no authority to "the group" or to "others", so, Bill won't be able to read it.

3.4. Example of changing an IFS objects permissions

To demonstrate the use of the chmod() API, we'll create a simple command that you can use to change the permissions on an IFS object.

Our command will need to know the path name of the IFS object, and the permissions to be assigned for the Owner, the Group and for everyone else. Our command source will look like this:

| CMD | PROMPT('Change File Mode') |
|------|--|
| PARM | KWD(OBJ) TYPE(*CHAR) LEN(640) MIN(1) + |
| | PROMPT('Object') |
| PARM | KWD(USER) TYPE(*CHAR) LEN(10) RSTD(*YES) + |
| | DFT(*NONE) VALUES(*NONE *R *RW *RX *RWX + |
| | *W *WX *X) PROMPT('Owner permissions') |

PARM KWD(GROUP) TYPE(*CHAR) LEN(10) RSTD(*YES) +
DFT(*NONE) VALUES(*NONE *R *RW *RX *RWX +
*W *WX *X) PROMPT('Group Permissions')
PARM KWD(OTHER) TYPE(*CHAR) LEN(10) RSTD(*YES) +
DFT(*NONE) VALUES(*NONE *R *RW *RX *RWX +
*W *WX *X) PROMPT('Others Permissions')

And the program that processes the command will look like this. Compile it, run it, laugh, cry, let it become a part of your being.

```
* CH3PERM: Example changing an IFS object's permissions
  (From Chap 3)
 *
 *
 * To compile:
    CRTBNDRPG CH3PERM SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
 *
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D Path
                 S
                               640A
D UserPerm
                S
                               10A
                               10A
D GroupPerm
                S
D OtherPerm
                S
                                10A
D Mode
                 S
                               10I 0
 ** Warning: call this program from the command. If you call
         it directly, because "Path" is larger than 32 bytes.
**
 * *
          See http://fag.midrange.com/data/cache/70.html
 * *
С
      *entry
                    plist
                                            Path
                    parm
С
                    parm
                                            UserPerm
С
                                            GroupPerm
С
                    parm
С
                    parm
                                            OtherPerm
С
                    eval
                             Mode = 0
C* Calculate desired user permissions:
                   if
                             scan('R': UserPerm) > 0
С
                    eval
                             Mode = Mode + S_IRUSR
С
                    endif
С
                    if
                             %scan('W': UserPerm) > 0
С
С
                    eval
                             Mode = Mode + S_IWUSR
                    endif
С
                    if
                             %scan('X': UserPerm) > 0
С
                    eval
                             Mode = Mode + S IXUSR
С
                    endif
С
C* Calculate desired group permissions:
                   if
                             %scan('R': GroupPerm) > 0
С
```

```
Mode = Mode + S_IRGRP
С
                     eval
С
                     endif
                     if
                               %scan('W': GroupPerm) > 0
С
                     eval
                               Mode = Mode + S_IWGRP
С
                     endif
С
                     if
                               scan('X': GroupPerm) > 0
С
                     eval
                               Mode = Mode + S IXGRP
C
                     endif
С
C* Calculate desired permissions for everyone else:
                     if
                               \$scan('R': OtherPerm) > 0
С
                     eval
                               Mode = Mode + S_IROTH
С
                     endif
С
                     if
                               %scan('W': OtherPerm) > 0
С
                     eval
                               Mode = Mode + S_IWOTH
С
                     endif
С
                     if
С
                               \$scan('X': OtherPerm) > 0
С
                     eval
                               Mode = Mode + S IXOTH
                     endif
С
C* Change the file's access mode:
                     if
                               chmod(%trimr(path): Mode) < 0</pre>
С
С
                     callp
                               die(%str(strerror(errno)))
                     endif
С
                               *inlr = *on
С
                     eval
 /DEFINE ERRNO_LOAD_PROCEDURE
 /COPY IFSEBOOK/QRPGLESRC, ERRNO_H
```

3.5. Retrieving Stream File Stats

Sometimes its useful to be able to look up information about a stream file. Information such as the file's size, access permissions, and the time it was last modified are all available using the stat() API.

Here is the C-language prototype for the stat() API:

```
int① stat(const char *path②, struct stat *buf③)
```

- The return value is an integer. Possible values are 0 which indicate success, or -1 if an error occurred.
- This is the path of the IFS object that you wish to get information about. This argument is a C-style string, so we use the options(*string) keyword in RPG so that the system will convert it to C's format for us.
- Here it wants a pointer to a stat data structure. This will be easy code in the prototype: we just define it as a pointer. However, we will also need to create a data structure in the same format as the stat data structure in C, which I'll explain below.

Here is the corresponding RPG prototype:

| D | stat | PR | 10I | 0 | <pre>ExtProc('stat')</pre> |
|---|------|----|-----|---|-----------------------------------|
| D | path | | * | | <pre>value options(*string)</pre> |
| D | buf | | * | | value |

In C, when you define a data structure, you first define the layout of the data structure. In this layout you list the subfields and their data types. Then, you declare variables that use this layout.

So, in that C-language prototype above, it tells us that buf is a pointer to any variable which uses the "stat layout".

Here is the definition for "struct stat" (the "stat layout") from the C language header member:

| struct stat { | | | | |
|----------------|-------------------------|-----|----------------------------------|------|
| mode_t | st_mode; | /* | File mode | */ |
| ino_t | st_ino; | /* | File serial number | */ |
| nlink_t | st_nlink; | /* | Number of links | */ |
| uid_t | st_uid; | /* | User ID of the owner of file | */ |
| gid_t | st_gid; | /* | Group ID of the group of file | */ |
| off_t | st_size; | /* | For regular files, the file | |
| | | * | size in bytes | */ |
| time_t | st_atime; | /* | Time of last access | */ |
| time_t | st_mtime; | /* | Time of last data modification | */ |
| time_t | st_ctime; | /* | Time of last file status change | */ |
| dev_t | st_dev; | /* | ID of device containing file | */ |
| size_t | st_blksize; | /* | Size of a block of the file | */ |
| unsigned long | st_allocsize; | /* | Allocation size of the file | */ |
| qp0l_objtype_t | st_objtype; | /* | AS/400 object type | */ |
| unsigned short | <pre>st_codepage;</pre> | /* | Object data codepage | */ |
| char | st_reserved1[62 | 2]; | /* Reserved | */ |
| unsigned int | st_ino_gen_id | /* | file serial number generation ic | 1 */ |
| }; | | | | |

The first line simply tells us that this is a structure definition called "stat".

The remaining lines, except for the last one, are a simple list of subfields, and their data types. For example, the last subfield in this data structure is called "st_ino_gen_id" and it is an unsigned integer.

To duplicate this in RPG, what we'll do is create a normal RPG data structure. But then, we'll base that structure on a pointer. Then, when we want to use the structure in our programs, we'll simply use the LIKE() keyword to declare character strings of the same size, and move the pointer so that we can reference the subfields. (Don't worry, if that's not clear to you, yet. I'll give you an example shortly that will help you understand.)

Also, you may have noticed that both the structure definition and the API are called "stat". That's not a problem in C since structure definitions use a separate name space from procedure calls. However, it is a problem in RPG. So, we'll call our data structure "statds" instead of "stat". That way, the name won't conflict with the name of the API.

Here is the RPG definition of the stat data structure:

| D | p_statds | S | * | |
|---|----------|----|-------|-----------------|
| D | statds | DS | | BASED(p_statds) |
| D | st_mode | | 10U C |) |
| D | st_ino | | 10U C |) |
| D | st_nlink | | 5U C |) |
| D | st_pad | | 2A | |

| D | st_uid | 10U | 0 |
|---|---------------|-----|---|
| D | st_gid | 10U | 0 |
| D | st_size | 101 | 0 |
| D | st_atime | 10I | 0 |
| D | st_mtime | 10I | 0 |
| D | st_ctime | 101 | 0 |
| D | st_dev | 10U | 0 |
| D | st_blksize | 10U | 0 |
| D | st_allocsize | 10U | 0 |
| D | st_objtype | 12A | |
| D | st_codepage | 5U | 0 |
| D | st_reserved1 | 62A | |
| D | st_ino_gen_id | 10U | 0 |
| | | | |

Now, when we call stat() in RPG, we'll do something like this:

```
D MyStat
                S
                                  like(statds)
D MySize
                S
                            10I 0
* get stat info into "MyStat":
С
                 if stat('/path/to/file.txt':
                                %addr(mystat)) < 0
С
                 callp EscErrno(errno)
С
                 endif
С
* move structure to overlay the "mystat" info:
                eval p_statds = %addr(mystat)
С
* read the file's size into MySize:
С
       eval MySize = st_size
```

3.6. Adding a *SAME option to the permission changer

In our last sample project, we created a program that assigned new access permissions to an IFS object. Now, let's update that example to allow "*SAME" to be specified.

This would, for example, allow you to change the permissions for the owner, without affecting the permissions for the group or anyone else.

To do that, what we'll do is retrieve the file's mode using stat(). Then, we'll preserve the bits from the original mode where *SAME is specified.

Here's the updated command source:

| CMD | PROMPT('Change File Mode') |
|------|---|
| PARM | KWD(OBJ) TYPE(*CHAR) LEN(640) MIN(1) + |
| | PROMPT('Object') |
| PARM | <pre>KWD(USER) TYPE(*CHAR) LEN(10) RSTD(*YES) +</pre> |
| | DFT(*SAME) VALUES(*NONE *R *RW *RX *RWX + |
| | *W *WX *X *SAME) PROMPT('Owner permissions') |

```
PARM KWD(GROUP) TYPE(*CHAR) LEN(10) RSTD(*YES) +
DFT(*SAME) VALUES(*NONE *R *RW *RX *RWX +
*W *WX *X *SAME) PROMPT('Group Permissions')
PARM KWD(OTHER) TYPE(*CHAR) LEN(10) RSTD(*YES) +
DFT(*SAME) VALUES(*NONE *R *RW *RX *RWX +
*W *WX *X *SAME) PROMPT('Others Permissions')
```

And here's the new RPG source:

```
* CH3PERM2: Example of changing permissions of an IFS object w/*SAME
 * (From Chap 3)
* To compile:
   CRTBNDRPG CH3PERM2 SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
*
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D Path
                 S
                              640A
D UserPerm
                 S
                               10A
D GroupPerm
                S
                               10A
D OtherPerm
                S
                              10A
D MyStat
                S
                                     like(statds)
D
                 DS
D Mode
                               10I 0
D CharMode3
                                1A overlay (Mode:3)
D CharMode4
                                   overlay(Mode:4)
                                1A
 ** Warning: call this program from the command. If you call
 * *
         it directly, because "Path" is larger than 32 bytes.
 * *
         See http://faq.midrange.com/data/cache/70.html
 **
С
     *entry
                   plist
                                           Path
                   parm
С
С
                   parm
                                           UserPerm
                   parm
                                           GroupPerm
С
                                           OtherPerm
С
                   parm
C* Retrieve current file mode:
                   if
                        stat(%trimr(path): %addr(mystat)) < 0</pre>
С
                   callp
                             die(%str(strerror(errno)))
С
                   endif
С
                   eval
                             p_statds = %addr(mystat)
С
                   eval
                            Mode = st_mode
С
C* Calculate desired user permissions:
                   if UserPerm <> '*SAME'
С
```

| _ | 1 | |
|---------------------|------------------|---|
| С | bitoff bitoff | x'FF' CharMode3 x'CO' CharMode4 |
| С | LIOII | x'CO' CharMode4 |
| С | if | <pre>%scan('R': UserPerm) > 0</pre> |
| c | eval | Mode = Mode + S IRUSR |
| c | endif | Hode - Hode + 5_1105K |
| c | if | <pre>%scan('W': UserPerm) > 0</pre> |
| c | eval | Mode = Mode + S_IWUSR |
| c | endif | |
| c | if | <pre>%scan('X': UserPerm) > 0</pre> |
| c | eval | Mode = Mode + S IXUSR |
| C | endif | ···· |
| | | |
| С | endif | |
| | | |
| C* Calculate desire | ed group pe | ermissions: |
| С | if | GroupPerm <> '*SAME' |
| | | |
| С | bitoff | x'38' CharMode4 |
| | | |
| С | if | scan('R': GroupPerm) > 0 |
| С | eval | Mode = Mode + S_IRGRP |
| С | endif | |
| С | if | <pre>%scan('W': GroupPerm) > 0</pre> |
| С | eval | Mode = Mode + S_IWGRP |
| С | endif | |
| С | if | <pre>%scan('X': GroupPerm) > 0</pre> |
| С | eval | Mode = Mode + S_IXGRP |
| С | endif | |
| | | |
| С | endif | |
| C+ Calculate desire | ad narmissi | ons for everyone else: |
| C Calculate desil | if | OtherPerm <> '*SAME' |
| C | ± ± | |
| С | bitoff | x'07' CharMode4 |
| - | | |
| С | if | scan('R': OtherPerm) > 0 |
| С | eval | Mode = Mode + S_IROTH |
| С | endif | |
| С | if | scan('W': OtherPerm) > 0 |
| С | eval | Mode = Mode + S_IWOTH |
| С | endif | |
| С | if | scan('X': OtherPerm) > 0 |
| С | eval | Mode = Mode + S_IXOTH |
| С | endif | |
| | | |
| С | endif | |
| | _ | |
| C* Change the file | | |
| С | if | <pre>chmod(%trimr(path): Mode) <</pre> |
| С | callp | die(%str(strerror(errno))) |
| С | endif | |

```
c eval *inlr = *on
/DEFINE ERRNO_LOAD_PROCEDURE
/COPY IFSEBOOK/QRPGLESRC,ERRNO_H
```

Note: We're using RPG's BITOFF operation to turn the bits from the original off before we set them. This ensures that they end up with the correct values, even if they were previously set to something.

3.7. Deleting IFS objects

You may have noticed that the IBM commands for working with the IFS frequently use the term "link." For example, the WRKLNK command ("Work with Links") is used to browse the IFS. The RMVLNK command ("Remove Link") is used to delete stream files.

You may be wondering "What's a link?"

To understand this, you need to make the distinction between the data that's stored in the file, and the file's name which shows up in the directory.

The data itself is called the "file". The name which you find in a directory is just a way of referring to that data. In essence, it's a link to the file.

In fact, it's possible to have more than one link to the same data. When that happens, the same file data may appear in more than one directory and/or under more than one different file name. Each one is considered a separate link, even though it's the same data.

When you remove a link to a stream file, the system will first remove the file name from the directory, and then it will check if this was the last link to the file. If the deleted link was the last link, the file's data will also be removed.

The API to delete a link is called "unlink()". And it's C-language prototype looks like this:

int unlink(const char *path)

Can't be much simpler than that, can it? It accepts only one parameter, and it's a null-terminated character string. It returns an integer, which will be a 0 if the API was successful or a -1 if it failed.

Here's the corresponding RPG prototype:

```
D unlink PR 10I 0 ExtProc('unlink')
D path * Value options(*string)
```

So, if you wanted to delete the stream file called "littlepgm.com", you'd write code that looks like this:

```
c if unlink('/ifstest/littlepgm.com') < 0
c callp EscErrno(errno)
c endif
```

3.8. Renaming IFS objects

It's easy to change the name of an IFS object. The system gives us a rename() API which can be called to do that.

The rename() API operates on the file's link, rather than on the file's data. This means that doing a rename() is much more efficient than copying the file, and deleting the old copy! All it has to change is the link, not the data.

Here is both the C and RPG prototypes for the rename() API. Once again, I won't bore you with the details of translating from C to RPG, since it is very straightforward on this API.

| | int rename(const | char | *old, | const | cha | *ne | w) |
|---|------------------|------|-------|-------|-------|-------|---------------------------------|
| | | | | | | | |
| D | rename | PR | | 101 | E O 1 | ExtPr | <pre>oc('Qp0lRenameKeep')</pre> |
| D | old | | | × | ÷ ۲ | /alue | options(*string) |
| D | new | | | * | ÷ ۲ | /alue | options(*string) |

One detail that should be noted is that the external procedure name isn't "rename" as you might expect, it's "Qp0lRenameKeep". The reason for this is that there are actually two different rename APIs. The difference between them is what happens when the "new name" is already the name of another IFS object.

If you're calling Qp0lRenameKeep, and the "new name" already exists, the system will return an error, protecting you from accidentally deleting the existing file.

If, instead, you call Qp0lRenameUnlink, the system will unlink the existing filename, and then proceed with the renaming. I never use this option because I feel it's safer, and more intuitive, to use Qp0lRenameKeep. I can always call unlink() beforehand if I really want to unlink an existing file.

Calling rename() from an RPG program is easy. Just do something like this:

```
c if rename('/ifstest/oldsmellyfile.dat':

c '/ifstest/newshinyfile.dat') < 0

c callp EscErrno(errno)

c endif
```

3.9. Example of renaming and deleting IFS objects

To demonstrate the use of the unlink() and rename() APIs, here's a sample program that can be used to either rename or delete a stream file.

What it does is ask for two parameters. The first is the current pathname of an IFS object. The second parameter is either the new pathname, or the special value *DELETE.

If *DELETE is specified, our program will bring up a window asking for confirmation before actually calling the unlink() API.

Here's the command source:

| CMD | <pre>PROMPT('Rename or Delete an IFS object')</pre> |
|------|---|
| PARM | KWD(OLD) TYPE(*CHAR) LEN(640) + |
| | <pre>PROMPT('Original (OLD) Object Name')</pre> |
| PARM | KWD(NEW) TYPE(*CHAR) LEN(640) + |

CHOICE('Character or *DELETE') + PROMPT('New Object Name or *DELETE')

Here's the DDS for the Window that it pops up:

А DSPSIZ(24 80 *DS3) Δ R DUMMYREC А ASSUME 1 2'' A А R RENDELS1 WINDOW(9 30 6 20) А 2 1'File:' А SCFILE 14 027 А 3 1'Size:' А А SCSIZE 10Y 00 3 7EDTCDE(L) A 5 1'Really? Delete it?' SCREALLY 1 I 520 Α

Here's the RPG code that makes it all work:

```
* CH3RENDEL: Example of deleting/renaming objects in the IFS
  (From Chap 3)
 *
 * To compile:
   CRTBNDRPG CH3RENDEL SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
 *
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
FCH3RENDELSCF E
                             WORKSTN
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
                                     'ABCDEFGHIJKLMNOPORSTUVWXYZ'
D upper
                 С
D lower
                 С
                                     'abcdefghijklmnopqrstuvwxyz'
D Path
                S
                             640A
               S
D NewPath
                              640A
D LowerNewPath
               S
                              640A
D MyStat
                S
                                    like(statds)
                              10I O
D Len
                S
D Pos
                 S
                               10I 0
 ** Warning: call this program from the command. If you call
         it directly, because "Path" is larger than 32 bytes.
 * *
         See http://faq.midrange.com/data/cache/70.html
 * *
 * *
                   plist
С
      *entry
                                           Path
                   parm
С
                                          NewPath
                   parm
С
                                          LowerNewPath
     upper:lower xlate
                           NewPath
С
```

```
С
               if
                       LowerNewPath = '*delete'
                      KillIt
С
               exsr
               else
С
                      NewIdentity
               exsr
С
               endif
С
               eval
                     *inlr = *on
С
C* Kill off the file (Delete it from the IFS)
CSR KillIt
              begsr
C*-----
C* Retrieve current file stats:
               if stat(%trimr(path): %addr(mystat)) < 0
С
С
               callp
                      die(%str(strerror(errno)))
С
               endif
C* Get file size from stats
             eval p_statds = %addr(mystat)
С
               eval
                      scSize = st_size
С
C* Strip directory names from front of pathname:
               eval Len = %len(%trimr(path))
С
С
               eval
                      Pos = Len
               dow
                      Pos > 0
С
               if
                       %subst(path:Pos:1) = '/'
С
               leave
С
               endif
С
               eval
                      Pos = Pos -1
С
               enddo
С
                    Pos<Len and %subst(path:Pos:1) = '/'
С
               if
               eval
                      scFile = %subst(path:Pos+1)
С
С
               else
С
               eval
                     scFile = path
               endif
С
C* Ask user if he/she REALLY wants to delete it?
               exfmt RENDELS1
С
C* Then ignore his choice and delete it anyway.
C* (just kidding)
С
               if
                      scReally = 'Y'
С
               if
                      unlink(%trimr(path)) < 0
                      die(%str(strerror(errno)))
               callp
С
С
               endif
               endif
С
C*-----
CSR
               endsr
```

/DEFINE ERRNO_LOAD_PROCEDURE
/COPY IFSEBOOK/QRPGLESRC,ERRNO_H

Chapter 4. Accessing stream files randomly

In chapter 2, we wrote data to the file at the start, and then read it from the start of the file. We always read the file sequentially. In this chapter, I'll show you how to read the file randomly.

4.1. Positioning to a given point in the file

In a standard physical file (one without key fields) you can position by record number using the SETLL and SETGT operations. This makes it possible to access a file "randomly" (or, in other words, you don't have to read through the file sequentially, you can "jump around".)

For a stream file, to read the file randomly, you use the "lseek()" API. However, since stream files are not organized into records by themselves, the lseek() API doesn't seek to a record number. Instead, you give it an "offset" which indicates a number of bytes, rather than records, to jump to.

Here is the C-language prototype for the lseek() API, followed by the corresponding RPG prototype:

| of | f_t lseek(int | fildes, | off_t | offset, | , | int whence) |
|------------|---------------|---------|-------|---------|---|-----------------------------|
| O D | lseek | PR | | 101 | 0 | <pre>ExtProc('lseek')</pre> |
| 0 D | fildes | | | 10I | 0 | value |
| ØD | offset | | | 10I | 0 | value |
| 4 D | whence | | | 10I | 0 | value |
| | | | | | | |

- The return value will be the new offset from the beginning of the file if lseek() was successful, otherwise it will be -1, and errno will be set.
- This is just the file descriptor of the stream file that you've opened.
- Here you put the offset. The offset is the number of bytes to move forward in the file from the point that's specified in the next parameter. If you wish to move backward instead of forward, you can specify a negative number. You can also specify zero if you want to move to exactly the point given in the next parameter.
- The "whence" parameter specifies where you want to move to, or start counting your offset from. It can be one of three named constants:
 - **SEEK_SET** means that the offset will be from the beginning of the file. An offset of 0 would be the very first byte in the file.
 - **SEEK_CUR** means that the offset will be from the current position in the file. For example, if you wanted to re-read the last 5 bytes, you could code **SEEK_CUR** with an offset of -5
 - SEEK_END means that the offset will be from the end of the file.

Now that I've told you about the named constants, it'd probably be a good idea to add them to our /copy member, eh?

| D | SEEK_SET | С | CONST(0) |
|---|----------|---|----------|
| D | SEEK_CUR | С | CONST(1) |
| D | SEEK_END | С | CONST(2) |

Here's a sample of jumping to the end of the file in RPG:

```
c if lseek(fd: 0: SEEK_END) < 0
c callp EscErrno(errno)
c endif
```

How about jumping to the 157th byte from the start of the file?

```
c if lseek(fd:157:SEEK_SET) < 0
c callp EscErrno(errno)
c endif
```

Or, if we're already at byte 157, we could easily seek forward to byte 167, like this:

| С | if | <pre>lseek(fd: 10: SEEK_CUR) < 0</pre> |
|---|-------|---|
| С | callp | EscErrno(errno) |
| С | endif | |

4.2. Example of using lseek() to jump around the file

Okay, here's a real example of using lseek() to jump around. Look over the code, and see if you can tell what it's going to do, then compile and run it.

```
* CH4RANDOM: Example of random access to an IFS object
*
  (From Chap 3)
 *
 * To compile:
   CRTBNDRPG CH3RANDOM SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
 *
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D fd
                S
                              10I 0
                              10I O
D err
                S
               S
D wrdata
                               48A
D rddata
                S
                               22A
D ShowMe
                S
                              48A varying
                   eval
                          fd = open('/ifstest/ch2_test.dat':
С
                                    O_WRONLY+O_CREAT+O_TRUNC:
С
С
                                    S_IRUSR+S_IWUSR+S_IRGRP)
                   if
                             fd < 0
С
                   callp
                             die('open(): ' + %str(strerror(errno)))
С
                   endif
С
```

```
C* Write some data
```

```
wrdata = 'THE QUICK BROWN FOX JUMP' +
С
                    eval
С
                                       'ED OVER THE LAZY GIRAFFE'
                    if
                              write(fd: %addr(wrdata): %size(wrdata))<1</pre>
С
                    eval
                              err = errno
С
                    callp
                              close(fd)
С
                    callp
                              die('write(): ' + %str(strerror(errno)))
С
                    endif
C
                    callp
                            close(fd)
С
                    eval
                              fd = open('/ifstest/ch2_test.dat':
С
                                     O_RDONLY)
С
                              fd < 0
                   if
С
                    callp
                              die('open(): ' + %str(strerror(errno)))
С
                    endif
С
                              (\text{ShowMe}) = 0
                    eval
С
C* Read the first 16 bytes
С
                    callp
                             read(fd: %addr(rddata): 16)
С
                    eval
                              ShowMe = ShowMe + %subst(rddata:1:16)
C* Jump to byte 41 of the file
C* and read 7 bytes
С
                    if
                              lseek(fd: 41: SEEK_SET) < 0</pre>
                    callp
                              die('lseek(): ' + %str(strerror(errno)))
С
С
                    endif
С
                    callp
                            read(fd: %addr(rddata): 7)
                    eval
                            ShowMe = ShowMe + %subst(rddata:1:7)
С
C* Jump to byte 19 of the file
C* and read 22 bytes
                    if
                              lseek(fd: 19: SEEK_SET) < 0</pre>
С
                              die('lseek(): ' + %str(strerror(errno)))
С
                    callp
                    endif
С
С
                    callp
                            read(fd: %addr(rddata): 22)
С
                    eval
                              ShowMe = ShowMe + %subst(rddata:1:22)
C* Jump to byte 16 of the file
C* and read 3 bytes
                    if
                              lseek(fd: 16: SEEK_SET) < 0</pre>
С
С
                    callp
                              die('lseek(): ' + %str(strerror(errno)))
С
                    endif
                    callp
                              read(fd: %addr(rddata): 3)
С
С
                    eval
                              ShowMe = ShowMe + %subst(rddata:1:3)
                          close(fd)
С
                    callp
C* Show what we read
                                            ShowMe
                    dsply
С
                    eval *inlr = *on
С
```

4.3. Organizing a stream file into records

Stream files are not record-based, but rather are just a collection of bytes. However, when you're working with records in your favorite physical file, what are they? Nothing more than a fixed-length bunch of bytes, right?

Theoretically, if we wrote fixed-length chunks of data to a stream file, and called those chunks "records", then we could use lseek() to jump to the start of each record, and read it just like a non-keyed physical file!

But, why would you ever do that? After all, the DB2/400 physical files are much more efficient, aren't they? Well, yes. But, let's say our file was going to be read directly by a PC program... aha! The PC program probably doesn't understand how to access the physical file, but it sure knows how to read a stream file!

Calculating the offset where a record in a stream file starts should be pretty easy. If we know, for example, that a record is 68 bytes long, and we want to jump to the 10th record in a file, all we have to do is multiply, right? Well, not exactly. If the 10th record is at position 680, then that would mean that the first record would be at $1 \ge 68$, or position 68. But, actually, the first record should be at offset 0, since that's the start of the file.

So, the formula for finding the offset for a start of a record is always: Offset = (RecordNo - 1) x RecordLength

4.4. Calculating number of records in a file

Figuring out how many records are in a stream file that has been organized into records is also quite simple. We just take the size of the file and divide it by the record length.

To simplify this slightly, I'm going to introduce another new API call. The fstat() API.

Here are the C and RPG prototypes:

```
int fstat(int fildes, struct stat *buf)
D fstat PR 10I 0 ExtProc('fstat')
D fildes 10I 0 value
D buf * value
```

I'm not going to explain the details of fstat(), since it's exactly like stat(), which we covered earlier.

The only difference between fstat() and the stat() API that we discussed in chapter 3, is that the fstat() API operates on a file descriptor, whereas the stat() API operates on a path name.

In other words, you use fstat() on a file that you've already opened, and you use stat() on a file that you don't need to open.

Once we've called fstat(), we can use the st_size subfield of the statds data structure to find the size of the stream file, and then divide it by our record length to find out how many records we have, like this:

c if fstat(fd: %addr(mystat)) < 0 c callp die('fstat(): ' + %str(strerror(err)))

```
c endif
c eval p_statds = %addr(mystat)
c eval numrec = st_size / record_len
```

4.5. Example of reading/writing/updating records in a stream file

This example will create a stream file that contains some demonstration records. It will then show you how to look up the records, update some of the information in them, and even search through the file to find them.

Here's the code. Once again, read the code over and see if you can figure out what it does. Then, run the program and see if you're right!

```
* CH4FIXED: Example of fixed-length records in an IFS file
  (From Chap 4)
 *
 * To compile:
    CRTBNDRPG CH4FIXED SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
 *
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D dsRecord
               DS
                             10S 0
D PartNo
                              10I 0
D Quantity
D
   UnitOfMeas
                               ЗA
  Price
                               7P 2
D
D Description
                              50A
D fd
                S
                             10I O
                 S
D err
                              10I 0
               S
D MyStat
                                    like(statds)
               S
                              10I 0
D recno
D NumRec
               S
                              10I 0
D SaveRec
                              10I 0
                S
                              10I 0
D Pos
                S
                          MakeFile
С
                   exsr
С
                   eval
                            fd = open('/ifstest/ch4_records': O_RDWR)
                   if
                            fd < 0
С
                   callp
                            die('open(): ' + %str(strerror(errno)))
С
                   endif
С
                   exsr
                            UpdateEx
С
                            SearchEx
С
                   exsr
```

| С | callp | close(fd) |
|---|-------|-------------|
| С | eval | *inlr = *on |

| C**** | ****** | ******** |
|--------------|---------------|---|
| CSR MakeFile | begsr | |
| с* | eval | <pre>fd = open('/ifstest/ch4_records':</pre> |
| с | | O_WRONLY+O_CREAT+O_TRUNC: |
| С | | <pre>S_IRUSR+S_IWUSR+S_IRGRP)</pre> |
| С | if | fd < 0 |
| с | callp | <pre>die('open(): ' + %str(strerror(errno)))</pre> |
| С | endif | |
| с | eval | PartNo = 5001 |
| С | eval | Quantity = 14 |
| С | eval | UnitOfMeas = 'BOX' |
| С | eval | Price = 7.95 |
| С | eval | Description = 'BLUE WIDGETS' |
| С | callp | write(fd:%addr(dsRecord):%size(dsRecord)) |
| с | eval | PartNo = 5002 |
| С | eval | Quantity = 6 |
| С | eval | UnitOfMeas = 'BOX' |
| c | eval eval | Price = 3.95 |
| c | callp | <pre>Description = 'RAINBOW SUSPENDERS' write(fd:%addr(dsRecord):%size(dsRecord))</pre> |
| C | Callp | witte(iu.%auui(uskecoiu).%size(uskecoiu)) |
| С | eval | PartNo = 5003 |
| С | eval | Quantity = 19 |
| С | eval | UnitOfMeas = 'SEA' |
| С | eval | Price = 29.95 |
| С | eval | Description = 'RED BICYCLE SEATS' |
| С | callp | write(fd:%addr(dsRecord):%size(dsRecord)) |
| с | eval | PartNo = 5004 |
| С | eval | Quantity = 8 |
| С | eval | UnitOfMeas = 'ITM' Price = 93512.80 |
| c | eval | |
| c | eval callp | <pre>Description = 'REALLY EXPENSIVE ITEMS' write(fd:%addr(dsRecord):%size(dsRecord))</pre> |
| | Callp | witte(iu.%auui(uskecoiu).%size(uskecoiu)) |
| с | eval | PartNo = 5005 |
| С | eval | Quantity = 414 |
| С | eval | UnitOfMeas = 'BAT' |
| С | eval | Price = 11.41 |
| С | eval | Description = 'BATS IN THE BELFRY' |
| С | callp | write(fd:%addr(dsRecord):%size(dsRecord)) |

```
PartNo = 5006
                      Quantity = 125
              eval
С
               eval
                      UnitOfMeas = 'BOX'
С
                      Price = 1.23
               eval
С
                      Description = 'KATES OLD SHOES'
               eval
С
               callp
                      write(fd:%addr(dsRecord):%size(dsRecord))
C
               callp
C
                     close(fd)
C*-----
CSR
               endsr
C* This demonstrates updating our fixed-length record file:
CSR UpdateEx
             begsr
C *-----
C* Someone bought a box of suspenders, and we want to change
C* the quantity:
C* The suspenders are in record number 2, so get them now:
               eval
                     recno = 2
С
                      pos = %size(dsRecord) * (recno-1)
               eval
С
               callp
                      lseek(fd: pos: SEEK_SET)
С
                      read(fd: %addr(dsRecord):%size(dsRecord))
С
               callp
               eval Quantity = Quantity - 1
callp lseek(fd: pos: SEEK_SET)
С
С
               callp
                      write(fd:%addr(dsRecord):%size(dsRecord))
С
C* Kate's shoes need to move faster, put them on sale!
               eval
                      recno = 6
С
                     pos = %size(dsRecord) * (recno-1)
               eval
С
                     lseek(fd: pos: SEEK_SET)
               callp
С
С
               callp
                     read(fd: %addr(dsRecord):%size(dsRecord))
                      Price = 0.25
С
               eval
С
               callp
                      lseek(fd: pos: SEEK_SET)
              callp
                      write(fd:%addr(dsRecord):%size(dsRecord))
С
C *-----
CSR
               endsr
C* This demonstrates searching for a record in the file:
CSR
   SearchEx
              begsr
C *-----
C* GASP! I can't remember the record number for Bats!
C* The part number was 5005, let's search for it
              if fstat(fd: %addr(mystat)) < 0</pre>
С
С
               eval
                      err = errno
```

С

eval

| С | callp | close(fd) |
|-----|--------|---|
| С | callp | <pre>die('fstat(): ' + %str(strerror(err)))</pre> |
| С | endif | |
| | | |
| С | eval | p_statds = %addr(mystat) |
| С | eval | numrec = st_size / %size(dsRecord) |
| С | eval | SaveRec = -1 |
| | | |
| С | for | recno = 1 to numrec |
| С | eval | pos = %size(dsRecord) * (recno-1) |
| С | callp | lseek(fd: pos: SEEK_SET) |
| С | callp | read(fd: %addr(PartNo): %size(PartNo)) |
| С | if | PartNo = 5005 |
| С | eval | SaveRec = recno |
| С | leave | |
| С | endif | |
| С | endfor | |
| | | |
| С | if | SaveRec = -1 |
| С | callp | close(fd) |
| С | callp | die('Part no 5005 not found!') |
| С | endif | |
| C* | | |
| CSR | endsr | |

/DEFINE ERRNO_LOAD_PROCEDURE /COPY IFSEBOOK/QRPGLESRC,ERRNO_H

Chapter 5. Text files

5.1. How do text files work?

As I mentioned at the start of this eBook, a stream file is simply a long string of bytes. How that string of bytes is used is up to the software that reads and writes the file.

One of the most common ways of organizing the data in a stream file is called a "text file." Text files are made up of human-readable text organized into variable-length records called "lines." Each line is intended to be printed on a single row of a display or a page of paper.

Text files are important to use because they are so widely used. This list is just a sample of the many places text files are used:

- · Nearly all printers accept text files as the their primary input format.
- All source code for PC programming languages, including C, C++, Visual Basic, Java, Perl and many others, is stored in a text file.
- Text files are the basis for more advanced file formats. HTML, XML, CSV and many other formats are text files with additional formatting added.
- All Internet e-mail is transmitted in text format. Even when you send pictures, movies or sound, they are converted to something that follows the rules of a text file before sending, and converted back after they're received.
- Nearly every major operating system comes with tools that work with text files. Windows comes with the
 "Notepad" program that edits text files. MS-DOS came with the "edit" and "edlin" programs. Unix comes with
 many tools for working with text, vi, ex, ed, grep, sed, awk, and more! Our familiar OS/400 even comes with
 commands such as "EDTF" and "DSPF" which are used to edit and display text files.

In order to make it easy to store variable-length lines, a special character is written to signify the end of a line. This character is called, appropriately, "end-of-line character" or more commonly the "new line character."

Unfortunately, not everyone agrees on what character should be used as the new line character. On Unix systems, they use the "LF" (line feed) character. On Macintosh systems, the "CR" (carriage return) character is used to indicate a new line, and for Microsoft Windows, they use the two characters in combination, CR followed by LF.

In our examples, we will use the Windows convention of "CRLF" since it is the most widely used. Changing your code to use just CR or just LF is easy enough to do if you need to work with one of the other formats.

5.2. Writing text data to a stream file

As mentioned previously, some human-readable text, followed by the CR and LF characters, will be viewed as a line of text. Consider the following code snippet:

| D CRLF | С | const(x'0d25') |
|--------|------|---|
| D text | S | 14a |
| | | |
| С | eval | <pre>text = 'Hello there.' + CRLF</pre> |

If I whip out my handy-dandy EBCDIC chart, I see that x'0D' is the EBCDIC code for carriage return, and x'25' is the EBCDIC code for line feed. Therefore, the variable called "text" above contains one line of text.

Consider this, more complicated, example:

| D CRLF C const(x'0d25') | |
|---------------------------------------|--------|
| D text S 500 | |
| | |
| c eval text = 'Hello there.' + CRLF + | CRLF + |
| c 'Nice day for a cruise.' + C | RLF + |
| c 'Maybe I"ll buy a yacht!' | |

Think about this: How many lines are stored in the variable called "text", above? Be careful before you answer, it's a trick question.

The answer is three and a half. Why? The first CRLF ends the first line, so the first line will read "Hello there!" in the text file. The next line ends when we encounter the next CRLF, which happens right away! That means the second line is blank. The third line says "Nice day for a cruise." and ends with another CRLF. The rest of the text, however, does not end with a CRLF. It's just part of a line.

All we have to do to put that text into a text file is call write(). This code would do the job:

c callp write(fd: %addr(text): c %len(%trimr(text)))

One small problem: If we let it be written like this, our text file would be invalid. Why? Because, remember, we left off the CRLF at the end of the last line.

To make sure that we don't make that mistake again, why not create some helper routines? What we'll do is create a service program, called IFSTEXTR4 and put our text file routines into it. Then, we'll use it for all the text files we write.

The routine to write a line will be simple. All it needs to do is accept parameters (just like write() does) for the descriptor, the text, and the text length. Then, we'll just stick a CRLF into the file after that.

Here's our IFSTEXTR4 service program, with the writeline() procedure:

```
** Service program to assist in creating TEXT files in the IFS
* *
H NOMAIN OPTION(*NOSHOWCPY: *SRCSTMT)
D/copy ifsebook/qrpglesrc, ifsio_h
D/copy ifsebook/qrpglesrc, ifstext_h
The concept here is very simple:
    1) Write the data passed to us into the stream file.
     2) Add the end of line characters.
P writeline
             В
                            export
            ΡI
D writeline
                       10I 0
D
  fd
                       10I 0 value
```

```
* value
D
  text
D
   len
                            10I 0 value
D rcl
              S
                            10I 0
                            10I 0
D rc2
              S
                S
D eol
                             2A
C* write the text provided
                 if
                        len > 0
С
                         rc1 = write(fd: text: len)
                 eval
С
                 if
                          rc1 < 1
С
                 return rc1
С
                  endif
С
                  endif
С
C* then add the end-of-line chars
                 eval eol = x' 0d25'
С
С
                  eval
                         rc2 = write(fd: %addr(eol): 2)
                         rc2 < 1
С
                 if
                 return
                          rc2
С
                  endif
С
                 return rc1 + rc2
С
                Е
Ρ
```

And then, we'll also need a prototype. That's what the /copy line for the "IFSTEXT_H" member is for. We'll put our prototypes in that member, so they can be easily accessed from other programs.

Here's the IFSTEXT_H member, so far:

```
/if defined(IFSTEXT_H)
 /eof
 /endif
 /define IFSTEXT_H
D writeline
                            10I 0
             PR
                             10I 0 value
D
  fd
D
   text
                              * value
                             10I 0 value
D
   len
```

Don't compile it, yet! We're also going to add a routine for reading text files.

5.3. Reading text data from a stream file

We know what a line of text looks like now. We also know how to write lines to disk. What we don't know, yet, is how to read them.

The big difference between reading lines and writing lines is that when you write the data, you already know how long the line needs to be. But when you read, you don't. You won't know how long it is until you've found the CRLF sequence in the text!

Now, we could solve that by reading one byte from disk at a time, until one of them turned out to be the new line sequence, but that would not run efficiently, because disk hardware is designed to read from disk in larger chunks. So, what we'll do is read a whole buffer of data, and then parse that data looking for our new line sequence.

We'll save any characters in the buffer that occur after the new line, so that we can use them as the start of our next line of text.

The RPG code that I came up with looks like this:

| P readline | В | export |
|-----------------|-----------------|---|
| D readline | PI | 10I 0 |
| D fd | | 10I 0 value |
| D text | | * value |
| D maxlen | | 10I 0 value |
| | | |
| D rdbuf | S | 1024A static |
| D rdpos | S | 10I O static |
| D rdlen | S | 10I O static |
| | | |
| D p_retstr | S | * |
| D RetStr | S | 32766A based(p_retstr) |
| D len | S | 101 0 |
| | 1 | |
| С | eval | len = 0 |
| С | eval | p_retstr = text |
| С | eval | <pre>%subst(RetStr:1:MaxLen) = *blanks</pre> |
| С | dow | 1 = 1 |
| C | aow | 1 — 1 |
| C* Load the bu | ffer | |
| С | if | rdpos>=rdlen |
| С | eval | rdpos = 0 |
| С | eval | rdlen=read(fd:%addr(rdbuf):%size(rdbuf)) |
| | | |
| С | if | rdlen < 1 |
| С | return | -1 |
| С | endif | |
| С | endif | |
| | | |
| C* Is this the | | |
| С | eval | rdpos = rdpos + 1 |
| С | if | <pre>%subst(rdbuf:rdpos:1) = x'25'</pre> |
| С | return | len |
| С | endif | |
| C. Otherwise | add it to the i | tout string |
| C* Otherwise, a | if | <pre>subst(rdbuf:rdpos:1) <> x'0d'</pre> |
| c | 1 L | and len<>maxlen |
| c | eval | len = len + 1 |
| c | eval | <pre>%subst(retstr:len:1) =</pre> |
| C | Eval | """" - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - """ - "" |

| C | 2 | | | <pre>%subst(rdbuf:rdpos:1)</pre> |
|---|---|---|--------|----------------------------------|
| C | 2 | | endif | |
| | | | | |
| C | | | enddo | |
| | | | | |
| C | 2 | | return | len |
| E |) | Е | | |
| | | | | |

Add that routine to the IFSTEXTR4 service program, and then add this code to the prototypes in the IFSTEXT_H member:

| D | readline | PR | 101 | 0 |
|---|----------|----|-----|---------|
| D | fd | | 101 | 0 value |
| D | text | | * | value |
| D | maxlen | | 101 | 0 value |

One more thing: Because we're writing a service program, and we want our code to be as useful as possible, we're going to write binder source that tells the CRTSRVPGM how other programs can bind to us.

If you haven't done this before, don't worry about it. It's really, really simple. It's just a list of what gets exported. Here's the entire code of our binding source:

```
STRPGMEXP
EXPORT SYMBOL(WRITELINE)
EXPORT SYMBOL(READLINE)
ENDPGMEXP
```

See? Very simple. Put that code into a member called "IFSTEXTR4" in a source file called QSRVSRC. Now let's compile it:

```
CRTRPGMOD IFSTEXTR4 SRCFILE(IFSEBOOK/QRPGLESRC) DBGVIEW(*LIST)
CRTSRVPGM IFSTEXTR4 EXPORT(*SRCFILE) SRCFILE(IFSEBOOK/QSRVSRC) TEXT('IFS Text file service program')
```

Finally, to make it easy to compile the programs that use this service program, we'll create a binding directory. This only involves running two commands:

```
CRTBNDDIR IFSEBOOK/IFSTEXT TEXT('IFS Text binding directory')
ADDBNDDIRE BNDDIR(IFSEBOOK/IFSTEXT) OBJ((IFSTEXTR4))
```

5.4. Example of writing and reading text files

Here's an example that uses our new service program. What it does is it creates a text file. Then it brings up the OS/400 text file editor to let you make changes to it. Finally, it reads back the text file, and displays the first 52 bytes of each line using the DSPLY op-code.

```
* CH5TEXT: Example of creating/reading a text file in the IFS
```

```
* (From Chap 5)
*
* To compile:
* CRTBNDRPG CH5TEXT SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D Cmd
             PR
                              ExtPgm('QCMDEXC')
D command
                        200A const
D len
                         15P 5 const
            S
S
                          10I O
D fd
                        100A
D line
D len
             S
                         10I O
D msg S
                         52A
               exsr MakeFile
С
                       EditFile
               exsr
С
С
               exsr
                       ShowFile
                eval
                        *inlr = *on
С
```

| C* 10 | rite some text | to a text file | |
|-------|---------------------------|----------------|--|
| C*** | * * * * * * * * * * * * * | ***** | |
| CSR | MakeFile | begsr | |
| C* | | | |

| C* | | |
|----|-------|---|
| С | eval | <pre>fd = open('/ifstest/ch5_file.txt':</pre> |
| С | | O_TRUNC+O_CREAT+O_WRONLY: |
| С | | S_IWUSR+S_IRUSR+S_IRGRP+S_IROTH) |
| С | if | fd < 0 |
| С | callp | die('open(): ' + %str(strerror(errno))) |
| С | endif | |
| | | |
| С | eval | line = 'Dear Cousin,' |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | | |
| С | eval | line = $'$ $'$ |
| С | eval | len = 0 |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | | |
| С | eval | line = $'I$ love the way you make $'$ + |
| С | | ' cheese fondue.' |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | | |
| С | eval | line = $'$ $'$ |
| С | eval | len = 0 |
| | | |

| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
|---------|-------|--|
| С | eval | line = 'Thank you for being so cheesy!' |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | eval | line = ' ' |
| С | eval | len = 0 |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | eval | <pre>line = 'Sincerely,'</pre> |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | eval | line = ' Richard M. Nixon' |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| c C* | callp | close(fd) |
| CSR | endsr | |

```
C* Read file, line by line, and dsply what fits
C* (DSPLY has a lousy 52-byte max... blech)
CSR ShowFile
             begsr
C*-----
             eval fd = open('/ifstest/ch5_file.txt':
С
                       O_RDONLY)
С
             if fd < 0
callp die('open(): ' + %str(strerror(errno)))</pre>
С
С
С
             endif
                  readline(fd: %addr(line): %size(line))>=0
Msg = line
С
             dow
             eval
С
С
   Msg
            dsply
             enddo
С
             callp close(fd)
С
```

```
c eval Msg = 'Press ENTER to continue'
c dsply Msg
C*-----
CSR endsr
/DEFINE ERRNO_LOAD_PROCEDURE
```

/COPY IFSEBOOK/QRPGLESRC, ERRNO_H

Try it out by compiling it and running it. Add some text and/or remove some text from the text file, and notice that the program will read it correctly.

5.5. Using code pages with text files

So, now we've created some text files, but there's one small problem. They're all in EBCDIC! This means that although we can read them using OS/400, they're utterly useless on the PC.

Now, we could call an API like QDCXLATE or iconv() to convert the text to ASCII before we write it to disk, but then our OS/400 programs wouldn't be able to work with it!

This problem can be easily solved, because the IFS allows us to assign a code page to our stream file. There are many code pages used to support all of the various characters used all around the world.

For our examples, we will use code page 37, which represents EBCDIC in the United States where I live, and code page 819 which represents ASCII here. If you're interested in using other code pages, please see the National Language Support manual which is published by IBM, and included in your Softcopy Library or Information Center.

The O_CODEPAGE flag on the open() API is used to assign a code page to a text file when the file is created.

So, when we want to assign a code page, we do this:

| С | callp | <pre>unlink('/ifstest/somefile.txt')</pre> |
|---|-------|---|
| с | eval | <pre>fd = open('/ifstest/somefile.txt':</pre> |
| С | | O_CREAT+O_WRONLY+O_CODEPAGE: |
| С | | mode: 819) |

Note that a new code page is assigned only if a new file is created. Therefore, if we really want to make sure that it's going to be assigned, we delete the file first.

The O_TEXTDATA flag on the open() API is used to tell the API that we're working with text data, so we want the system to translate the data for us. Note that it only does the translation if the file already exists, and is assigned a different code page than the one that's associated with our job.

To open the file so that the translation takes place, we'd open it like this:

c eval fd = open('/ifstest/somefile.txt': c O_RDWR+O_TEXTDATA) Now when we read or write any data to that file, it will automatically be translated to or from the code page that was assigned to it.

5.6. Example of writing & creating an ASCII text file

Here we will take the same example that we created earlier in the chapter, and add code page support to it. The data that we write will actually be translated to ASCII for us.

Note also that the EDTF command still works, even though it's now in ASCII. The EDTF command uses the code page that we assigned, and understands that it needs to translate it as well!

You can also take the stream file, transfer it to your PC, and open it up with Notepad, Wordpad or even Word. Because the files are in ASCII, they can be used almost anywhere!

```
* CH5ASCII: Example of text file in ASCII mode
* (From Chap 5)
* To compile:
  CRTBNDRPG CH5ASCII SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
*
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D Cmd
              ΡR
                              ExtPqm ('QCMDEXC')
  command
D
                        200A const
D
  len
                         15P 5 const
D fd
             S
                         10I 0
             S
D line
                        100A
             S
D len
                         10I O
             S
                         52A
D msq
D err
             S
                         10I 0
                       MakeFile
С
               exsr
                       EditFile
С
                exsr
                        ShowFile
С
                exsr
                        *inlr = *on
C
                eval
C* Write some text to a text file
CSR
   MakeFile
               begsr
C*-----
C* Make sure we don't have an old file that might be in the way
C* (ENOENT means it didnt exist to begin with)
               if unlink('/ifstest/ch5_file.txt') < 0</pre>
С
С
                eval
                       err = errno
```

```
С
                    if
                              err <> ENOENT
С
                    callp
                              die('unlink(): ' + %str(strerror(err)))
                    endif
С
                    endif
С
C* Create a new file, and assign it a code page of 819:
                    eval
                               fd = open('/ifstest/ch5_file.txt':
C
                                    O_CREAT+O_WRONLY+O_CODEPAGE:
С
                                    S_IWUSR+S_IRUSR+S_IRGRP+S_IROTH:
С
                                    819)
С
                    if
                               fd < 0
С
                              die('open(): ' + %str(strerror(errno)))
                    callp
С
                    endif
С
                    callp
                              close(fd)
С
C* Now re-open the file in text mode. Since it was assigned a
C* code page of 819, and we're opening it in text mode, OS/400
C* will automatically translate to/from ASCII for us.
С
                    eval
                              fd = open('/ifstest/ch5_file.txt':
С
                                         O_WRONLY+O_TEXTDATA)
С
                    if
                               fd < 0
                    callp
                              die('open(): ' + %str(strerror(errno)))
С
                    endif
С
С
                    eval
                              line = 'Dear Cousin,'
                    eval
                              len = %len(%trimr(line))
С
С
                    callp
                              writeline(fd: %addr(line): len)
                              line = ' '
                    eval
С
                    eval
                              len = 0
С
                    callp
                              writeline(fd: %addr(line): len)
С
                              line = 'I love the way you make' +
                    eval
С
                                ' cheese fondue.'
С
                    eval
                              len = %len(%trimr(line))
С
С
                    callp
                              writeline(fd: %addr(line): len)
                    eval
                              line = ' '
С
                    eval
                              len = 0
С
                    callp
                              writeline(fd: %addr(line): len)
С
                    eval
                              line = 'Thank you for being so cheesy!'
C
С
                    eval
                              len = %len(%trimr(line))
                              writeline(fd: %addr(line): len)
                    callp
С
С
                    eval
                              line = ' '
                    eval
                              len = 0
С
С
                    callp
                              writeline(fd: %addr(line): len)
                              line = 'Sincerely,'
С
                    eval
                              len = %len(%trimr(line))
                    eval
С
                              writeline(fd: %addr(line): len)
                    callp
С
```

```
eval line = ' Richard M. Nixon'
eval len = %len(%trimr(line))
С
С
            callp
                  writeline(fd: %addr(line): len)
С
                  close(fd)
C
            callp
C*-----
CSR
            endsr
C* Call the OS/400 text editor, and let the user change the
C* text around.
CSR EditFile begsr
C*-----
            callp cmd('EDTF STMF("/ifstest/' +
С
                            'ch5_file.txt")': 200)
С
C *-----
CSR
            endsr
C* Read file, line by line, and dsply what fits
C* (DSPLY has a lousy 52-byte max... blech)
CSR ShowFile
            begsr
C *-----
            eval
                  fd = open('/ifstest/ch5_file.txt':
С
                      O_RDONLY+O_TEXTDATA)
С
            if fd < 0
С
            callp
                  die('open(): ' + %str(strerror(errno)))
С
            endif
С
                  readline(fd: %addr(line): %size(line))>=0
С
            dow
            eval
                  Msg = line
С
С
  Msg
            dsply
С
            enddo
            callp close(fd)
С
                 Msg = 'Press ENTER to continue'
С
            eval
                           Msg
C
            dsply
C+-----
CSR
            endsr
```

/DEFINE ERRNO_LOAD_PROCEDURE /COPY IFSEBOOK/QRPGLESRC,ERRNO_H

5.7. Example of a report in ASCII

Now, let's do something a little more practical as an example. Have you ever had someone ask if a report can be sent to their PC, instead of printed? Here's an example of doing that with a stream file.

Since the files that a report would typically be printed off of are not a standard part of OS/400, I decided to make a report by listing the objects in a given library. The library will be a parameter.

It might be a good exercise for you to create your own report. Use mine as a sample, but take a report that's commonly used in your company, and convert it to write a text file as well as the printer output.

Anyway, here's the code:

```
* CH5LIBLIST: Example of a report in ASCII text format
*
  (From Chap 5)
* To compile:
   CRTBNDRPG CH5LIBLIST SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
FOADSPOBJ IF E
                     K DISK
                             USROPN
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D Cmd
             PR
                             ExtPqm('QCMDEXC')
D
   command
                        200A const
D
   len
                        15P 5 const
D FmtDate
             PR
                        10A
   mmddyy
                        6A
D
                            const
             S
                        10I 0
D fd
                        100A
D line
             S
                        10I 0
D len
             S
                        10I 0
D LineNo
            S
              plist
С
    *entry
С
               parm
                                  MyLib
                                               10
C* Create a file containing the objects we wish to report
С
               callp
                       cmd('DSPOBJD OBJ('+%trim(MyLib)+'/*ALL)'+
С
                        ' OBJTYPE(*ALL) OUTPUT(*OUTFILE) ' +
С
                         ' OUTFILE (QTEMP/QADSPOBJ) ' +
                         ' OUTMBR(*FIRST *REPLACE)': 200)
С
C* Open the list of objects:
cmd('OVRDBF FILE(QADSPOBJ) TOFILE(' +
               callp
С
```

| с | | 'QTEMP/QADSPOBJ)': 200) |
|---------------------|--------------|--|
| С | open | QADSPOBJ |
| | | |
| | | ***** |
| C* Open a stream fi | | * |
| | | ***** |
| c | eval | <pre>fd = open('/ifstest/object_report.txt':</pre> |
| c | | S_IRWXU+S_IRWXG+S_IROTH: 819) |
| с | if | fd < 0 |
| c | callp | die('open(): ' + %str(strerror(errno))) |
| C | endif | |
| С | callp | close(fd) |
| | - | |
| С | eval | <pre>fd = open('/ifstest/object_report.txt':</pre> |
| С | | O_TEXTDATA+O_WRONLY) |
| С | if | fd < 0 |
| С | callp | <pre>die('open(): ' + %str(strerror(errno)))</pre> |
| С | endif | |
| | | |
| | | * |
| C* Create the repor | | |
| | | ************************************** |
| c c | exsr read | Heading QADSPOBJ |
| C | IEau | QAD3F0B0 |
| С | dow | not %eof(QADSPOBJ) |
| С | exsr | WriteObj |
| С | read | QADSPOBJ |
| С | enddo | |
| | | |
| | | |
| | | * |
| c* Clean up and exi | | |
| 0 | | ***** |
| С | - | close(fd) |
| с | | QADSPOBJ cmd('DLTOVR FILE(QADSPOBJ)': 50) |
| c | callp | cmd('DLTF QTEMP/QADSPOBJ': 50) |
| C | carth | |
| С | eval | *inlr = *on |
| | | |
| | | |
| C*================= | | |
| C* Write a heading | - | |
| C*============ | | |
| CSR Heading | | |
| C* | | |
| C* x'OC' = Form Fee | | |
| C | eval eval | line = x'Oc' len = 1 |
| c c | | <pre>len = 1 writeline(fd: %addr(line): len)</pre> |
| C | σαττρ | witcetine(id. oddar(tine). ien) |

| С | eval | line = 'Listing of objects in ' + |
|-------------------|-------|---|
| С | | %trim(MyLib) + ′ library′ |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | | |
| С | eval | line = *blanks |
| С | eval | len = 0 |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | eval | line (Object News) |
| С | | |
| С | eval | <pre>%subst(line: 15) = 'Object Type'</pre> |
| С | eval | |
| С | eval | <pre>%subst(line: 45) = 'Last Modified'</pre> |
| С | eval | |
| С | eval | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | eval | line = '' |
| С | | |
| С | eval | <pre>%subst(line: 15) = ''</pre> |
| С | eval | |
| С | eval | <pre>%subst(line: 45) = ''</pre> |
| С | eval | %subst(line: 60) = '' |
| С | eval | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | 1 | T de a Na |
| С | | LineNo = 5 |
| C* | | |
| csr | endsr | |
| | | |
| 0. | | |
| C *============== | | |

| C* Add an object to the report | | | | |
|--------------------------------|-------|--|--|--|
| Cx CSR WriteObj C* | begsr | | | |
| С | if | LineNo > 60 | | |
| С | exsr | Heading | | |
| С | endif | | | |
| | | | | |
| С | eval | Line = odObNm | | |
| С | eval | %subst(line: 15) = odobtp | | |
| С | eval | <pre>%subst(line: 30) = %editc(odobsz:'L')</pre> | | |
| С | eval | <pre>%subst(line: 45) = FmtDate(odldat)</pre> | | |
| С | eval | <pre>%subst(line: 60) = FmtDate(odudat)</pre> | | |
| С | eval | len = %len(%trimr(line)) | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| | | | | |
| С | eval | LineNo = LineNo + 1 | | |
| C * | | | | |
| csr | endsr | | | |
| | | | | |

* Format a date into human-readable YYYY-MM-DD format: P FmtDate В ΡI 10A D FmtDate 6A const D mmddyy 6 0 D Temp6 S D TempDate D S D Temp10 S 10A C* If date isn't a valid number, return *blanks С testn mmddyy 99 if *in99 = *offС С return *blanks endif С C* If date isn't a valid MMDDYY date, return *blanks Temp6 С move mmddyy Temp6 С *mdy test(de) С if %error С return *blanks endif С C* Convert date to ISO format, and return it. С *mdy move Temp6 TempDate С *iso move TempDate Temp10 Temp10 С return Ρ Е

/DEFINE ERRNO_LOAD_PROCEDURE
/COPY IFSEBOOK/QRPGLESRC,ERRNO_H

Chapter 6. Additional Text Formats

6.1. Comma Separated Values (CSV) Format

One type of text file that's commonly used is called "Comma Separated Values" (CSV) format. In CSV, you create a file, where each "record" in the file is a line of text, each line is broken up into "fields", and each field is separated by a comma.

CSV is useful for taking data from an OS/400 physical file and importing it into another program. Perhaps the most popular program is Microsoft Excel, but there are many others that will accept CSV as an input.

The OS/400 command called CPYTOIMPF can be used to create a stream file in CSV format without us having to write any code to do it. So, why are we doing this? One reason is that it helps you understand a little more about stream files. Another reason is that sometimes a little detail comes up that CPYTOIMPF does not support, and when that happens it's nice to have an alternative. Finally, there are times when your CSV file needs to contain data that requires program logic to generate.

6.2. Example of creating a CSV file

In chapter 5, we created a report in plain ASCII format that showed the objects in a library. Now, we will find the same data, and put it into a CSV-formatted stream file.

```
* CH6CSV: Example of a report in ASCII CSV format
   (From Chap 6)
 * To compile:
    CRTBNDRPG CH6CSV SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
 *
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
FQADSPOBJ IF
                E
                             K DISK
                                       USROPN
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D Cmd
                  ΡR
                                       ExtPgm('QCMDEXC')
D
    command
                                200A
                                      const
D
    len
                                 15P 5 const
D FmtDate
                  ΡR
                                 10A
    mmddyy
                                  6A
D
                                       const
D fd
                  S
                                 10I 0
D line
                  S
                                100A
D len
                  S
                                 10I 0
D LineNo
                  S
                                 10I 0
```

| С | *entry | plist | | | |
|--|---------------------------------------|---|--|--|--|
| С | | parm | MyLib 10 | | |
| C.**** | * * * * * * * * * * * * * * | * * * * * * * * * * * | * | | |
| | | | the objects we wish to report | | |
| C**** | * * * * * * * * * * * * * | - * * * * * * * * * * | | | |
| С | | callp | <pre>cmd('DSPOBJD OBJ('+%trim(MyLib)+'/*ALL)'+</pre> | | |
| С | | | <pre>' OBJTYPE(*ALL) OUTPUT(*OUTFILE) ' +</pre> | | |
| С | | | ' OUTFILE (QTEMP/QADSPOBJ) ' + | | |
| С | | | <pre>' OUTMBR(*FIRST *REPLACE)': 200)</pre> | | |
| C++++ | · · · · · · · · · · · · · · · · · · · | · • • • • • • • • • • • • • • • • • • • | * | | |
| - | en the list of | | | | |
| C**** | * * * * * * * * * * * * * | * * * * * * * * * * * | * | | |
| С | | callp | <pre>cmd('OVRDBF FILE(QADSPOBJ) TOFILE(' +</pre> | | |
| С | | | 'QTEMP/QADSPOBJ)': 200) | | |
| С | | open | QADSPOBJ | | |
| C.**** | * * * * * * * * * * * * * * * | * * * * * * * * * * * | * | | |
| | en a stream fi | | | | |
| C**** | * * * * * * * * * * * * * * | ********* | * | | |
| С | | eval | <pre>fd = open('/ifstest/object_report.csv':</pre> | | |
| С | | | O_CREAT+O_TRUNC+O_CODEPAGE+O_WRONLY: | | |
| С | | | S_IRWXU+S_IRWXG+S_IROTH: 819) | | |
| С | | if | fd < 0 | | |
| С | | callp | <pre>die('open(): ' + %str(strerror(errno)))</pre> | | |
| С | | endif | | | |
| С | | callp | close(fd) | | |
| С | | eval | <pre>fd = open('/ifstest/object_report.csv':</pre> | | |
| С | | | O_TEXTDATA+O_WRONLY) | | |
| С | | if | fd < 0 | | |
| С | | callp | die('open(): ' + %str(strerror(errno))) | | |
| С | | endif | | | |
| C**** | * * * * * * * * * * * * * * * | * * * * * * * * * * * | ***** | | |
| C* Cr | eate the repor | rt | | | |
| C**** | * * * * * * * * * * * * * | ********* | * | | |
| С | | read | QADSPOBJ | | |
| С | | dow | not %eof(QADSPOBJ) | | |
| c | | exsr | WriteObj | | |
| С | | read | QADSPOBJ | | |
| С | | enddo | 2 | | |
| - | | | | | |
| C • • | | | | | |
| C************************************* | | | | | |
| | ean up and exi | | * | | |
| - | ************ | callp | close(fd) | | |
| с с | | close | QADSPOBJ | | |
| c | | callp | cmd('DLTOVR FILE(QADSPOBJ)': 50) | | |
| c | | callp | cmd('DLTF_QTEMP/QADSPOBJ': 50) | | |
| <u> </u> | | 00115 | ······································ | | |

eval *inlr = *on

С

C* Add an object to the report CSR WriteObj begsr C*-----Line = '"' + %trim(odobnm) + '",' + С eval '"' + %trim(odobtp) + '",' + С %trim(%editc(odobsz:'L')) + ',' + С '"' + %trim(FmtDate(odldat)) + '",' + C '"' + %trim(FmtDate(odudat)) + '"' С len = %len(%trimr(line)) С eval writeline(fd: %addr(line): len) С callp C *----csr endsr * Format a date into human-readable YYYY-MM-DD format: P FmtDate В ΡI 10A D FmtDate D mmddyy 6A const S D Temp6 60 D TempDate S D S 10A D Temp10 C* If date isn't a valid number, return *blanks 99 С testn mmddyy if *in99 = *off С С return *blanks С endif C* If date isn't a valid MMDDYY date, return *blanks move mmddyy Temp6 С *mdy test(de) Temp6 С %error if С return *blanks С endif С C* Convert date to ISO format, and return it. *mdy move Temp6 TempDate С С *iso move TempDate Temp10 return Temp10 С Ρ Е

/DEFINE ERRNO_LOAD_PROCEDURE

/COPY IFSEBOOK/QRPGLESRC, ERRNO_H

See if you can figure out what that code is doing, then compile it and run it. You can open up the results using EDTF or Windows Notepad, and see what the output looks like.

Then, open up the same file using Microsoft Excel. Note that each field appears in a separate column in the spreadsheet. Now, you can use Excel's different capabilities to sort the file, or total up the columns, or whatever you'd like to do.

Next, you might try using this technique with one of your company's reports. Have you had a user ask if he could import the results of a report into Excel? This could be used as a way to do that!

6.3. HTML (web page) format

We already saw in Chapter 5 that text files could be opened in Microsoft Word. They don't have any fancy characteristics, however. No bold, no fonts, no underlines... Just plain boring text.

HTML is an easy way to add some of these extra bells and whistles. HTML stands Hypertext Markup Language and is the language that web pages are written in. Since HTML documents can be read by Microsoft Word, as well as web browsers, it is a useful format to be able to write data into.

Unfortunately, fully explaining the features of HTML would be far beyond the scope of this document. However, I will give you a quick idea of how it works.

HTML consists of "mark up tags". The concept is that you have this plain text document, and then you insert these tags to identify how parts of the document should be displayed.

Most HTML capabilities have a starting and an ending tag. For example, to mark something as "bold", you use the "b" tag. The starting tag looks like this: $\langle b \rangle$ and the ending tag is the same, except that it starts with a slash like this: $\langle b \rangle$. Everything placed in between these tags will appear as bold text.

```
This is normal. <b>This is bold.</b>
```

6.4. Example of creating an HTML file

This example will, once again, generate a report of the objects in library. However, this time we've used HTML to format the report.

Here's the code:

```
* CH5LIBLIST: Example of a report in HTML format
* (From Chap 6)
*
* To compile:
* CRTBNDRPG CH6HTML SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
*
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
FQADSPOBJ IF E K DISK USROPN
```

```
D/copy IFSEBOOK/ORPGLESRC, IFSIO H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D Cmd
            PR
                         ExtPqm('QCMDEXC')
D command
                     200A const.
D len
                     15P 5 const
D FmtDate
           PR
                     10A
D mmddyy
                     6A const
D fd
                     10I O
           S
Dline
           S
                    100A
D len
           S
                     10I 0
           plist
c *entry
С
             parm
                              MyLib
                                         10
C* Create a file containing the objects we wish to report
callp
                    cmd('DSPOBJD OBJ('+%trim(MyLib)+'/*ALL)'+
С
                     ' OBJTYPE(*ALL) OUTPUT(*OUTFILE) ' +
С
С
                      ' OUTFILE (OTEMP/OADSPOBJ) ' +
С
                      ' OUTMBR(*FIRST *REPLACE)': 200)
C* Open the list of objects:
callp cmd('OVRDBF FILE(QADSPOBJ) TOFILE(' +
C
                       'QTEMP/QADSPOBJ)': 200)
С
                    QADSPOBJ
С
             open
C* Open a stream file to write report to:
eval
                   fd = open('/ifstest/object_report.html':
С
                       O_CREAT+O_TRUNC+O_CODEPAGE+O_WRONLY:
С
                       S_IRWXU+S_IRWXG+S_IROTH: 819)
С
             if
                    fd < 0
С
                    die('open(): ' + %str(strerror(errno)))
С
             callp
С
             endif
             callp
                   close(fd)
С
С
             eval
                  fd = open('/ifstest/object_report.html':
                           O_TEXTDATA+O_WRONLY)
С
             if
С
                    fd < 0
             callp
                    die('open(): ' + %str(strerror(errno)))
С
             endif
С
```

C* Create the report

| С | exsr | Heading |
|---|-------|--------------------|
| С | read | QADSPOBJ |
| | | |
| С | dow | not %eof(QADSPOBJ) |
| С | exsr | WriteObj |
| С | read | QADSPOBJ |
| С | enddo | |
| | | |
| С | exsr | Footer |

| C************************************* | | | | |
|--|----------------------------------|---|--|--|
| с с с | callp close callp callp | close(fd) QADSPOBJ cmd('DLTOVR FILE(QADSPOBJ)': 50) cmd('DLTF QTEMP/QADSPOBJ': 50) | | |
| с | eval | <pre>*inlr = *on</pre> | | |

| CSR C* | Heading | begsr | |
|-----------|---------|-------|--|
| C | | eval | line = ' <html><head><title>'</td></tr><tr><td>С</td><td></td><td>eval</td><td>len = %len(%trimr(line))</td></tr><tr><td>С</td><td></td><td>callp</td><td><pre>writeline(fd: %addr(line): len)</pre></td></tr><tr><td>С</td><td></td><td>eval</td><td>line = 'Listing of objects in ' +</td></tr><tr><td>С</td><td></td><td></td><td>%trim(MyLib) + ' library'</td></tr><tr><td>С</td><td></td><td>eval</td><td><pre>len = %len(%trimr(line))</pre></td></tr><tr><td>С</td><td></td><td>callp</td><td><pre>writeline(fd: %addr(line): len)</pre></td></tr><tr><td>С</td><td></td><td>eval</td><td><pre>line = '</title></head>'</html> |
| С | | eval | <pre>len = %len(%trimr(line))</pre> |
| С | | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | | eval | line = ' <body>'</body> |
| С | | eval | len = %len(%trimr(line)) |
| С | | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | | eval | line = ' <h1><center>' +</center></h1> |
| С | | | 'Listing of objects in ' + |
| С | | | %trim(MyLib) + ′ library′ |
| С | | eval | <pre>len = %len(%trimr(line))</pre> |
| С | | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | | eval | line = ' <center>' +</center> |
| С | | | <pre>//</pre> |

| С | eval | <pre>len = %len(%trimr(line))</pre> | | |
|-----|-------|--|--|--|
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| | | | | |
| С | eval | line = $' '$ | | |
| С | eval | len = %len(%trimr(line)) | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| | 1 | | | |
| С | eval | <pre>line = 'Object Name'</pre> | | |
| С | eval | <pre>len = %len(%trimr(line))</pre> | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| с | eval | line = ' Object Type ' | | |
| | eval | | | |
| С | | <pre>len = %len(%trimr(line))</pre> | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| с | eval | line = ' Object Size ' | | |
| c | eval | <pre>len = %len(%trimr(line))</pre> | | |
| | | | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| С | eval | <pre>line = 'Last Modified'</pre> | | |
| С | eval | len = %len(%trimr(line)) | | |
| С | callp | writeline(fd: %addr(line): len) | | |
| | | | | |
| С | eval | line = ' Last Used ' | | |
| С | eval | len = %len(%trimr(line)) | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| | | | | |
| С | eval | line = $' '$ | | |
| С | eval | len = %len(%trimr(line)) | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | |
| C* | | | | |
| csr | endsr | | | |
| | | | | |

| C*==================================== | | | | | | |
|--|------------|--|--|--|--|--|
| C* Add an object to the report | | | | | | |
| 0 | eObj begsr | | | | | |
| C * | | | | | | |
| С | eval | line = $' '$ | | | | |
| С | eval | len = %len(%trimr(line)) | | | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | | | |
| | | | | | | |
| С | eval | line = '' +%Trim(odObNm) +' ' | | | | |
| С | eval | len = %len(%trimr(line)) | | | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | | | |
| | | | | | | |
| С | eval | line = ''+%Trim(odObTp)+'' | | | | |
| С | eval | len = %len(%trimr(line)) | | | | |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> | | | | |
| | | | | | | |
| С | eval | line = $' < td$ align=right> $' +$ | | | | |
| С | | <pre>%trim(%editc(od0bSz:'L')) + ''</pre> '' | | | | |

| С | eval | <pre>len = %len(%trimr(line))</pre> |
|-----|-------|--|
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| С | eval | line = $' < td$ align=center>' + |
| С | | %trim(FmtDate(odldat)) + '' |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| | | |
| С | eval | line = $' < td$ align=center> $'$ + |
| С | | <pre>%trim(FmtDate(odudat)) + ''</pre> |
| С | eval | len = %len(%trimr(line)) |
| С | callp | writeline(fd: %addr(line): len) |
| | | |
| С | eval | line = '' |
| С | eval | len = %len(%trimr(line)) |
| С | callp | <pre>writeline(fd: %addr(line): len)</pre> |
| C* | | |
| csr | endsr | |

C* Finish up the HTML page CSR Footer beqsr C *----eval line = '</center>'
eval len = %len(%trimr(line))
callp writeline(fd: %addr(line): len) С С С eval line = '</body></html>' С eval len = %len(%trimr(line)) С callp writeline(fd: %addr(line): len) С C+----csr endsr

```
*****
* Format a date into human-readable YYYY-MM-DD format:
*****
P FmtDate B
         ΡI
D FmtDate
                  10A
D mmddyy
                  6A const
        S
                  6 0
D Temp6
D TempDate
         S
                  D
D Temp10
         S
                  10A
C* If date isn't a valid number, return *blanks
           testn
                         mmddyy
                                     99
С
           if
                *in99 = *off
С
           return ' '
С
           endif
С
```

```
C* If date isn't a valid MMDDYY date, return *blanks
С
                   move
                            mmddyy
                                          Temp6
С
     *mdy
                  test(de)
                                          Temp6
С
                   if
                            %error
                            ' '
                   return
С
                   endif
С
C* Convert date to ISO format, and return it.
С
     *mdy
                 move
                            Temp6
                                          TempDate
     *iso
                            TempDate
                                          Temp10
С
                  move
С
                  return
                            Temp10
Ρ
                 Е
 /DEFINE ERRNO_LOAD_PROCEDURE
 /COPY IFSEBOOK/QRPGLESRC, ERRNO_H
```

Go ahead and run it, compile it, then bring up the resulting file using EDTF or the Windows Notepad. Then, try bringing the same file up in your web browser or in Word. Notice the affect of the various tags?

Chapter 7. Working with directories

7.1. How directories work

Back when we first discussed stream files, and path names, we talked about directories. We know that directories are similar to libraries in their ability to act as "containers" for stream files. In other words, stream files are stored inside directories in the IFS.

In addition to storing stream files inside a directory, most file systems also allow you to store directories inside a directory. A directory inside a directory is sometimes referred to as a "sub-directory." From a programmers perspective, directories and sub-directories behave exactly the same way, so we will simply refer to them all as "directories."

Each directory has a "mode" (access permission bits) just as the stream files do. Most file systems allow you to run chmod() to change the mode, just as you would with a file.

The mode of a directory acts slightly differently, however. If you have "read" permission to a directory, it means that you're allowed to see the list of files that it contains. If you have "write" permission, it means that you're able to add files, delete files, rename files, etc in the directory. If you have "execute" permission to a directory, it means that you're allowed to "search" the directory. Some programs, such as the "find" utility in QShell, will skip searching directories that you do not have execute authority to.

Just in case you missed it, let me stress this again: If you have "write" permission to a directory, you can add, delete, or rename files in that directory. Even though you don't have access to the file itself!

For example, let's say you created a directory called "scott". In that directory, you put a file called "dont_let_bob_read.txt". Let's say that the directory's mode allows read, write and execute to the owner, the group and everyone else. Let's say that the file gives read and write access to the owner, but nothing to the group or the world. Okay, now Bob sees his name on the file, he tries to open it with DSPF, but he can't. He can't read the file. So, he tries to delete it... He can! The file is gone.

Why can Bob do that? Because, technically, when you are adding, removing, or changing the names of stream files in a directory, you aren't modifying the file data. You're adding, removing or changing entries in the directory itself.

There are two "special" entries inside each directory, as well. They are "." (a single dot) and ".." (two dots). These signify the "current directory" and the "parent directory" respectively.

7.2. Creating directories

The API for creating a new directory (or sub-directory) in the IFS is called "mkdir()." This stands, appropriately, for "make directory."

Here are the C language and RPG prototypes for the mkdir() API. They're straightforward, so I will not bore you with the conversion details,

int mkdir(const char *path, mode_t mode)

| ÛD | mkdir | PR | <pre>10I 0 ExtProc('mkdir')</pre> |
|------------|-------|----|--|
| 2 D | path | | Value options(*string) |
| ØD | mode | | 10U 0 Value |

- mkdir() returns -1 if it failed, or 0 if it was successful
- The "path" parameter is where you specify the directory to be created. The format of the path is just like that specified in the open() API.
- The "mode" parameter is the access permissions to assign to the directory when it's created. You specify the mode using the same bit-flags that you used for the mode on the open() API.

For example, consider the following code:

```
C if mkdir('/ifstest/ShoeTongue':

C S_IRUSR+S_IWUSR+S_IXUSR+

C S_IRGRP+S_IWGRP+S_IXGRP+

C S_IROTH+S_IXGRP) < 0

C callp EscErrno(errno)

C endif
```

This code creates a new sub-directory called "ShoeTongue" inside the a directory called "/ifstest". It grants read, write and search permissions to the owner, read, write and search permissions to the primary group, and read and search permissions to everyone else.

In order for this command to exist, the "/ifstest" directory must already exist. If not, the command will fail with "path or directory not found!"

7.3. Removing directories

What can be created, can also be destroyed. The rmdir() "Remove Directory" API is used to delete a directory.

Here are the C and RPG prototypes for the rmdir() API.

```
int rmdir(const char *path)
D rmdir PR 10I 0 ExtProc('rmdir')
D path * value options(*string)
```

In order for a directory to be deleted, it must be empty. You cannot delete a directory that still has files in it. Here's a sample of deleting the ShoeTongue directory:

```
c if rmdir('/ifstest/ShoeTongue') < 0
c callp EscErrno(errno)
c endif
```

7.4. Switching your current directory

Just as there is a "current library" in the traditional file system, there is also a "current directory" in the IFS. When you supply a path name to an API, and that path name does not start with a slash character, it tells the API that you want to use the "current directory."

You can change your current directory by calling the chdir() "Change Directory" API.

The prototype for chdir() looks like this:

| i | .nt chdir(const | char | *path) | | | |
|---|-----------------|------|--------|-----|---|-----------------------------|
| D | chdir | PR | | 101 | 0 | <pre>ExtProc('chdir')</pre> |
| D | path | | | * | | Value Options(*string) |

Not much to it, is there? You just call chdir() with one argument, the path to change to. The chdir() API will return a -1 if it fails, or a 0 if it was successful.

In this example, we will create and delete the ShoeTongue directory, just as we did in the code examples for mkdir() and rmdir(). However, we will use a "relative" directory name this time. We will switch our current directory to /ifstest, and then we won't have to specify it on the mkdir() and rmdir() APIs.

| С | if | chdir('/ifstest') < 0 |
|---|-------|---------------------------|
| С | callp | EscErrno(errno) |
| С | endif | |
| С | if | mkdir('ShoeTongue': |
| С | | S_IRUSR+S_IWUSR+S_IXUSR+ |
| С | | S_IRGRP+S_IWGRP+S_IXGRP+ |
| С | | S_IROTH+S_IXOTH) < 0 |
| С | callp | EscErrno(errno) |
| С | endif | |
| С | if | rmdir('ShoeTongue') < 0 |
| С | callp | EscErrno(errno) |
| С | endif | |

7.5. Opening Directories

Now that we know how to create and delete directories, and change our current directory, we also need to know how to read the contents of a directory.

The process of reading directories in an RPG program will require you to use 3 different APIs. They are opendir(), which opens the directory, readdir() which reads the next entry from a directory and closedir() which closes the directory when you're finished.

The opendir() API is similar in some ways to the open() API. It accepts a parameter that tells the name of a directory to open, and it returns a handle that can be used to read through that directory.

Here is the C-language prototype for the opendir() API:

DIR *opendir(const char *dirname)

We know that the "DIR *" means that it returns a pointer to a "DIR". But what data type is a "DIR"? We could figure that out -- if you hunt through the C header members, you'd eventually find it. But, as it turns out, we don't need it! All we need to do with the return value from opendir() is pass it as a parameter to the readdir() and closedir() APIs. Therefore, we don't care what the pointer points to! We can just treat it as a pointer.

When this API fails, it will return a *NULL pointer, and we can then check error to find out which error occurred. Therefore, the prototype for the opendir() API in RPG looks like this:

| D | opendir | PR | * | EXTPROC('opendir') |
|---|---------|----|---|------------------------|
| D | dirname | | * | VALUE options(*string) |

When we want to open a directory, we simply pass the directory name as a parameter. For example, if our goal was to read the contents of the root directory of the IFS, we might write code that looks like this:

```
D d S *
c eval d = opendir('/')
c if d = *NULL
c callp EscErrno(errno)
c endif
```

7.6. Reading Directories

Once the directory has been opened, we will want to read it's contents. This is accomplished using the readdir() "Read Directory" API.

Here is the C language prototype for the readdir() API:

```
struct dirent *readdir(DIR *dirp)
```

The return value of readdir() is a pointer that points to a data structure in the "dirent" format. Back in the discussion of the stat() API, I explained that in C, you first define a data structure's format, and then you define structures that use that format. In this case, the format is called "dirent".

We will need to know how to define our own dirent data structures, since that's where the useful information from the directory comes from. However, for creating the RPG prototype, we really only needed to know that the return value is a pointer. So, here's the RPG prototype:

```
D readdir PR * EXTPROC('readdir')
D dirp * VALUE
```

To get useful information out of this API, as I mentioned above, we need to make our own dirent structure. The definition of this structure in C looks like this;

```
struct dirent {
```

```
d_reserved1[16]; /* Reserved
 Ochar
                                                                      */
 Qunsigned int
                 d fileno gen id; /* File number generation ID @A1C*/
 €ino_t
                  d_fileno;
                                    /* The file number of the file
                                                                      */
 4unsigned int
                 d_reclen;
                                   /* Length of this directory entry
                                     in bytes
                                                                     */
 Gint
                 d_reserved3;
                                    /* Reserved
                                                                      */
 6 char
                  d reserved4[8];
                                    /* Reserved
                                                                      * /
 0qlg_nls_t
                 d_nlsinfo;
                                    /* National Language Information
                                      about d name
                                                                     */
 Ounsigned int
                 d_namelen;
                                    /* Length of the name, in bytes
                                      excluding NULL terminator
                                                                     */
 Ochar
                  d_name[_QPOL_DIR_NAME]; /* Name...null terminated
                                                                      */
};
```

- The first subfield in the dirent data structure is simple. It's a 16-byte long character field marked as "reserved".
- Next, we have the file number generation ID. It's marked as an unsigned integer, which is equivalent to the RPG "10U 0" data type. This field probably won't be useful to you as an RPG programmer, so I won't explain what it contains.
- The file number is defined as a "ino_t". If we hunt through the C header members, we will find that ino_t is defined in the file QSYSINC/SYS, TYPES as an unsigned integer. Good, that's an RPG "10U 0". Again, this field isn't likely to be useful to an RPG programmer, so I won't explain it.
- The length of the directory entry is stored here, and it's marked as another unsigned integer. Lots of "10U 0" definitions for our structure, eh?
- This field is marked as "reserved" and is an integer, which we know is defined as "10I 0" in RPG.
- Yet another "reserved" field. This one is an 8-byte character field.
- The national language support information is stored here. This information allows us to determine what CCSID we should be using to display the filename, as well as what language it is in, and what country it is from.

If we hunt for "qlg_nls_t" in the C header members, we'll find it defined in QSYSINC/SYS, TYPES. It is defined to be a data structure, containing four more subfields, an integer containing the CCSID, a 2-byte field containing the country-id, a 3-byte field containing the language ID, and a 3-byte reserved field.

- This subfield contains the length of the filename that will be given in the next subfield. Since it is listed as an unsigned integer, we know that we need to make the RPG version use a "10U 0" variable.
- Finally, what we've been waiting for! The name of the file that this directory entry refers to. The constant
 "_QPOL_DIR_NAME" defines the maximum length that the file name can be. If we hunt through the C header
 members, we will find that this constant is set to be the number 640.

So, here's our RPG version of the dirent structure. Note that we've based it on a pointer. This is important, since the readdir() API allocates the memory for this structure, consequently, we need to be able to point our structure at it's memory.

| Dр | _dirent | S | * | |
|----|----------------|----|-------|----------------------------|
| Dd | lirent | ds | | <pre>based(p_dirent)</pre> |
| D | d_reserved1 | | 16A | |
| D | d_fileno_gen_: | id | | |
| D | | | 10U 0 | |
| | | | | |

| D | d_fileno | 10U | 0 | |
|---|-------------|------|---|-----------------------|
| D | d_reclen | 10U | 0 | |
| D | d_reserved3 | 101 | 0 | |
| D | d_reserved4 | 8A | | |
| D | d_nlsinfo | 12A | | |
| D | nls_ccsid | 101 | 0 | OVERLAY(d_nlsinfo:1) |
| D | nls_cntry | 2A | | OVERLAY(d_nlsinfo:5) |
| D | nls_lang | ЗA | | OVERLAY(d_nlsinfo:7) |
| D | nls_reserv | ЗA | | OVERLAY(d_nlsinfo:10) |
| D | d_namelen | 10U | 0 | |
| D | d_name | 640A | | |

Once we've added these to our IFSIO_H member, we'll be able to call the readdir() API like this:

| С | | | eval | <pre>p_dirent = readdir(d)</pre> |
|----|----|------------|------------|----------------------------------|
| С | | | dow | p_dirent <> *NULL |
| C* | do | whatever w | e like wit | h the contents |
| C* | of | the dirent | structure | , here |
| С | | | eval | <pre>p_dirent = readdir(d)</pre> |
| С | | | enddo | |

7.7. Closing an open directory

When we're done reading the directory, we need to close it. To do that, we will call the Close Directory API, closedir().

Here's the C and RPG prototypes for closedir():

```
int closedir(DIR *dirp)
D closedir PR 10I 0 EXTPROC('closedir')
D dirhandle * VALUE
```

Calling this API is quite simple, just pass it the variable that you received when you called opendir().

For example:

| С | if | closedir(d) < 0 |
|---|-------|-----------------|
| С | callp | EscErrno(errno) |
| С | endif | |

7.8. Example of reading a directory

In this simple example, we will open up our /ifstest directory that we've been using for all of our sample code, and we'll read the contents of it. For each entry in that directory, we'll use the DSPLY op-code to display the first 52 bytes of the file name.

```
* CH7READDIR: Example of reading a directory in the IFS
   (From Chap 7)
 *
 * To compile:
 *
    CRTBNDRPG CH7READDIR SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D dir
                  s
                                   *
                  S
D Msq
                                 52A
                              dir = opendir('/ifstest')
С
                    eval
                    if
                              dir = *NULL
С
С
                    callp
                              die('opendir(): '+%str(strerror(errno)))
                    endif
С
                    eval
                              p_dirent = readdir(dir)
С
                              p_dirent <> *NULL
                    dow
С
                    eval
                              Msg = %subst(d_name:1:d_namelen)
С
                    dsply
С
      msg
                    eval
                              p_dirent = readdir(dir)
С
                    enddo
С
                    callp
                              closedir(dir)
С
С
                    eval
                              Msg = 'Press ENTER to end'
                    dsply
                                             Msg
С
                    eval
                               *inlr = *on
С
 /DEFINE ERRNO_LOAD_PROCEDURE
 /COPY IFSEBOOK/QRPGLESRC, ERRNO_H
```

If you want to experiment with it, you can change the directory name to other directories, and run it again. Or, maybe have it always list your current directory, and use chdir() to change which directory it lists.

7.9. Example of making a DIR command for QSHELL

Here's a slightly more complex example of reading a directory. This program is designed to be run under the QShell utility that IBM provides.

QShell is intended to emulate a UNIX command prompt, and therefore it provides the UNIX "ls" command to view directories. But, what if you like the way directories look in the MS-DOS "dir" command? Well... we could write out own DIR command using the APIs, right? Let's try!

One of the interesting things about QSHELL, is that it provides us with 3 stream file descriptors, already opened, which we can use for input and output to the shell.

Descriptor #0 is referred to as "standard input", but we will not use it in our example.

Descriptor #1 is referred to as "standard output". It's the place to write our directory listing to. We should treat it like a text file, which is to say that we start a new line by sending a CRLF sequence.

Descriptor #2 is referred to as "standard error". Any error messages that we need to communicate to the user should be sent there.

There's a lot more that I could explain about QSHELL, but alas, it's a bit beyond the scope of this document to fully explain this environment...

We'll also need to send text to the screen listing whether or not a file is a directory, and also the file size, because these are done in MS-DOS. Fortunately, we can use the stat() API to retrieve these.

Finally, we'll need to report the date that's associated with each file in the directory. We can get that from the stat() API as well, but all the dates in stat() are listed in Universal Time Coordinated (UTC) and not in our time zone. The dates that appear, are actually listed as a number of seconds since the "epoch" of January 1st, 1970. So, to make all of this behave as we want it to, we've got our work cut out for us. Still, hopefully when you see the code, you'll understand what's going on!

Here's the code:

```
* CH7QSHDIR: More complex example of reading a dir in the IFS
   We try to make our output look like the MS-DOS "DIR" command
   (From Chap 7)
 *
 * To compile:
    CRTBNDRPG CH7QSHDIR SRCFILE(xxx/QRPGLESRC) DBGVIEW(*LIST)
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D STDIN
                  С
                                       CONST(0)
                  С
D STDOUT
                                       CONST(1)
                  С
D STDERR
                                       CONST(2)
D S ISDIR
                  ΡR
                                  1N
D
   mode
                                 10U 0 value
D CEEUTCO
                  ΡR
                                       ExtProc('CEEUTCO')
                                 10I 0
D
   hours
                                 10I 0
D
   minutes
                                  8F
D
   seconds
D line
                               1024A
                  S
D len
                                 10I 0
                  s
```

```
S
D dir
                         *
D filename
            S
                      1024A varying
            S
D fnwithdir
                      2048A varying
D mystat
            S
                            like(statds)
D curr
            S
                     1024A
            S
D curdir
                      1024A varying
            S
                       10T 0
D dot
            S
                       10I 0
D notdot
D epoch
            S
                        Z inz(z'1970-01-01-00.00.00.000000')
            S
                        ЗA
D ext
            S
D filedate
                        8A
            S
D filetime
                        6A
D modtime
            S
                        Z
D mydate
            S
                        D
            S
                         Т
D mytime
            S
D shortfn
                        8A
            S
D size
                       13A
D worktime
            S
                        8A
D hours_utc
            S
                       10I O
D mins_utc
            S
                       10I 0
            S
D secs_utc
                        8F
                       10I 0
D utcoffset
            S
* Here's an example of what an MS-DOS directory listing
* looks like:
 *
  Directory of C:\WINDOWS
              <DIR> 10-24-00 10:28a .
 *
  .
* .. <DIR> 10-24-00 10:28a ..
* COMMAND <DIR> 05-08-00 11:54a COMMAND
 * VB INI 1,245 10-04-01 9:12p VB.INI
 * LOCALS~1
                        10-06-01 9:44p Local Settings
              <DIR>
               1,986 11-21-01 8:43p bddkl.drv
 * BDDKL DRV
* BPKPC DRV
                  3,234 11-21-01 8:43p bpkpc.drv
 * PILCIKK DRV
                  1,122 11-21-01 8:43p pilcikk.drv
* NEWRES~1 RC
                  1,440 12-12-01 2:02a newrestest.rc
              eval *inlr = *on
С
C* get the number of seconds between local
C* time an Universal Time Coordinated (UTC)
С
              callp(e) CEEUTCO(hours_utc: mins_utc: secs_utc)
С
              if
                     %error
               eval
                     utcoffset = 0
С
С
               else
              eval utcoffset = secs_utc
С
              endif
С
C* Use the getcwd() API to find out the
```

```
C* name of the current directory:
if
                         getcwd(%addr(curr): %size(curr)) = *NULL
С
                 eval
                         line = 'getcwd(): ' +
С
                                 %str(strerror(errno))
С
                 eval
                         len = %len(%trimr(line))
С
                 callp
                         writeline(STDERR: %addr(line): len)
C
                 return
С
                 endif
С
                 eval
                       curdir = %str(%addr(curr))
С
C* open the current directory:
eval
                         dir = opendir(curdir)
С
                 if
                         dir = *NULL
С
С
                 eval
                         line = 'opendir(): ' +
                                 %str(strerror(errno))
С
                 eval
                        len = %len(%trimr(line))
С
                 callp
                         writeline(STDERR: %addr(line): len)
С
                 return
С
                 endif
С
                         line = "
С
                 eval
                 eval
                         len = 0
С
С
                 callp
                         writeline(STDOUT: %addr(line): len)
                         line = ' Directory of ' + curdir
С
                 eval
                         len = %len(%trimr(line))
                 eval
С
                 callp
                         writeline(STDOUT: %addr(line): len)
С
                         line = "
                 eval
С
                          len = 0
                 eval
С
                         writeline(STDOUT: %addr(line): len)
                 callp
С
С
                 eval
                        p_statds = %addr(mystat)
                 eval
                         p_dirent = readdir(dir)
С
                 dow
                         p_dirent <> *NULL
С
                 eval
                         filename = %subst(d_name:1:d_namelen)
С
                 eval
                         fnwithdir = curdir + '/' + filename
С
                 if
                          stat(fnwithdir: %addr(mystat))=0
С
                         PrintFile
                 exsr
С
                 endif
С
                 eval
                        p_dirent = readdir(dir)
С
                 enddo
С
                 callp
                        closedir(dir)
С
```

 $\ensuremath{\mathsf{C}}\star$ For each file in the directory, print a line of info:

```
CSR
   PrintFile
               beqsr
C*-----
C* Separate into extension & short filename:
   ′.′
                       filename notdot
С
               check
                if
                        notdot = 0
С
                        ext = *blanks
                eval
C
                eval
                       shortfn = filename
С
                else
С
                eval dot = %scan('.': filename: notdot)
С
                if
                        dot > 0
С
                eval
                       ext = %subst(filename:dot+1)
С
                       shortfn = %subst(filename: 1: dot-1)
С
                eval
                else
С
                       ext = *blanks
С
                eval
                       shortfn = filename
                eval
С
                endif
С
С
                endif
C* Show size if this is not a directory:
                if
                        S_ISDIR(st_mode)
С
                       size = '<DIR>'
С
                eval
                else
С
                eval
                       size = %editc(st_size: 'K')
С
С
                endif
C* figure out date & time:
   epoch
                adddur st_atime:*S modtime
С
                adddur utcoffset:*S modtime
С
С
               move
                       modtime mydate
               move
                       modtime
                                  mytime
С
   *MDY-
                       mydate
                                  filedate
С
               move
   *USA
                        mytime
                                   worktime
               move
С
                eval
                        filetime=%subst(worktime:1:5) +
С
                               %subst(worktime:7:1)
С
C* and write it to QSH STDOUT:
                       line = shortfn + ' ' + ext + ' ' +
                eval
С
                          size + ' ' + filedate + ' ' +
С
                          filetime + ' ' + filename
С
                eval
                        len = %len(%trimr(line))
С
                        writeline(STDOUT: %addr(line): len)
C
                callp
C*-----
CSR
                endsr
* This tests a file mode to see if a file is a directory.
 * Here is the C code we're trying to duplicate:
     #define _S_IFDIR 0040000
 *
      #define S_ISDIR(mode) (((mode) & 0370000) == _S_IFDIR)
 *
```

*/

```
* 1) ((mode) & 0370000) takes the file's mode and performs a
       bitwise AND with the octal constant 0370000. In binary,
       that constant looks like: 00000000000000111110000000000
 +
       The effect of this code is to turn off all bits in the
       mode, except those marked with a '1' in the binary bitmask.
 *
 * 2) ((result of #1) == _S_IFDIR) What this does is compare
       the result of step 1, above with the _S_IFDIR, which
       is defined to be the octal constant 0040000. In decimal,
       that octal constant is 16384.
 P S ISDIR
                В
D S_ISDIR
                ΡI
                              1 N
                              10U 0 value
D
   mode
D
                DS
                               4U 0
D dirmode
                        1
D byte1
                        1
                               1 A
                              2A
D byte2
                        2
D byte3
                        3
                               ЗA
D byte4
                        4
                               4A
C* Turn off bits in the mode, as in step (1) above.
                            dirmode = mode
                  eval
С
                  bitoff
                            x'FF'
                                         byte1
С
С
                  bitoff
                            x'FE'
                                         byte2
                  bitoff
                            x'0F'
                                         byte3
С
                  bitoff
                                         bvte4
                            x'FF'
С
C* Compare the result to 0040000, and return true or false.
                  if
                            dirmode = 16384
С
                            *On
С
                  return
                  else
С
                            *Off
С
                  return
С
                  endif
Ρ
                Е
 /DEFINE ERRNO_LOAD_PROCEDURE
```

/COPY IFSEBOOK/QRPGLESRC, ERRNO_H

You'll need to run this from the QSHELL. If your system doesn't have QSHELL installed, you've got a choice. You can either try to modify my code so that it doesn't need QSHELL, and make it work, or you can install QSHELL. QSHELL should be on your OS/400 CD's, and you should be able to install it free-of-charge.

To run this, do the following:

- 1. If you haven't done so already, compile the program according to the instructions at the top of the code.
- 2. Start the QSHELL by typing: **STRQSH** at your OS/400 command prompt.
- 3. Run the command by typing: /QSYS.LIB/IFSEBOOK.LIB/CH7QSHDIR.PGM at the QSHELL command prompt.

4. If that's too much to type, create a symbolic link to the program by typing:

```
cd /ifstest
```

```
ln -s /QSYS.LIB/IFSEBOOK.LIB/CH7QSHDIR.PGM dir
```

And then adding the /ifstest directory to your path by typing:

PATH=\${PATH}:/ifstest

5. Now you can just type **dir** to list the contents of the directory.

7.10. Example of Reading a directory recursively

One more program. This one shows how to read through directories "recursively."

As you know, a directory can contain sub-directories. Those sub-directories can contain more sub-directories. How can you write a program that will process them all, when you don't know how many levels deep it can go?

The answer is "recursion." Recursion is the ability for a sub-procedure to call itself. Each new call to the sub-procedure has it's own copies of the variables that it uses (unless they are global or static.) This is useful to us, since it means that we can write a sub-procedure that lists out the contents of a directory. When it finds a sub-directory in that directory, it can call itself with the sub-directory's name. The new copy of the sub-procedure will process that directory, again looking for more directories, etc.

It's a bit of a difficult concept to understand the first time you see it. It helps to think of it as a "call stack", like programs have. Each time the procedure calls itself, think of it adding a new entry to that stack. When the newly called procedure ends, it's entry on the stack is removed, and the copy that made the call continues where it left off.

Once again, we'll use QSHELL. But this time, instead of looking like an MS-DOS DIR command, we'll just print the filename to the screen. Hopefully, when you see what it's outputting, you'll understand how the recursion works.

```
* CH7RECURSE: Just in case that last example wasn't complicated
     enough!
 *
 *
   (From Chap 7)
 * To compile:
    CRTBNDRPG CH7RECURSE SRCFILE (xxx/QRPGLESRC) DBGVIEW (*LIST)
H DFTACTGRP(*NO) ACTGRP(*NEW) BNDDIR('QC2LE') BNDDIR('IFSTEXT')
D/copy IFSEBOOK/QRPGLESRC, IFSIO_H
D/copy IFSEBOOK/QRPGLESRC, ERRNO_H
D/copy IFSEBOOK/QRPGLESRC, IFSTEXT_H
D show_dir
                  PR
                                10T 0
D curdir
                              1024A varying const
                  С
D STDIN
                                      CONST(0)
D STDOUT
                  С
                                      CONST(1)
D STDERR
                  С
                                      CONST(2)
D S_ISDIR
                  PR
                                 1 N
D mode
                                10U 0 value
```

```
D curr
         S
                   1024A
D curdir
           S
                   1024A varying
D line
           S
                   1024A
D len
           S
                     10I O
           PR
                        ExtPgm('QCMDEXC')
D cmd
D command
                    200A const
D length
                     15P 5 const
            eval *inlr = *on
С
                   cmd('DLYJOB DLY(10)': 20)
С
             callp
C* Use the getcwd() API to find out the
C* name of the current directory:
C
             if
                   getcwd(%addr(curr): %size(curr)) = *NULL
                 line = 'getcwd(): ' +
             eval
С
С
                         %str(strerror(errno))
             eval len = %len(%trimr(line))
С
             callp writeline(STDERR: %addr(line): len)
С
             return
С
С
             endif
С
             eval
                   curdir = %str(%addr(curr))
C* Call our show_dir proc to show all
C* of the files (and subdirectories) in
C* the current directory.
С
             callp show_dir(curdir)
             return
С
* prints all of the files in the directory to STDOUT.
* if a subdirectory is found, this procedure will call
* itself (recursively) to process that directory.
*****
P show_dir
           В
D show_dir
           ΡI
                    10I O
D curdir
                   1024A varying const
         S
D mystat
                         like(statds)
                    *
           S
D dir
D line
                   1024A
           S
           S
                     10I O
D len
           S
D filename
                   1024A varying
                    1024A varying
D fnwithdir S
```

D err S 10I 0

| C****** | * * * * * * * * * * * | **** |
|---------------------|-----------------------|---|
| C* open the current | t directory | /: |
| C****** | | |
| С | eval | dir = opendir(curdir) |
| С | if | dir = *NULL |
| С | eval | err = errno |
| С | eval | <pre>line = 'opendir(): ' +</pre> |
| С | | %str(strerror(err)) + |
| С | - | <pre>', errno=' + %trim(%editc(err:'L'))</pre> |
| С | eval | <pre>len = %len(%trimr(line))</pre> |
| С | callp | <pre>writeline(STDERR: %addr(line): len)</pre> |
| С | if | err = EACCES |
| С | return | 0 |
| С | else | |
| С | return | -1 |
| С | endif | |
| С | endif | |
| |] | |
| С | eval | p_dirent = readdir(dir) |
| С | dow | p_dirent <> *NULL |
| | | |
| С | eval | filename = %subst(d_name:1:d_namelen) |
| С | eval | <pre>fnwithdir = curdir + '/' + filename</pre> |
| | | |
| C | if | filename<>'.' and filename<>'' |
| | | |
| С | if . | <pre>stat(fnwithdir: %addr(mystat))<0</pre> |
| С | eval | line = 'stat(): ' + |
| С | 2 | <pre>%str(strerror(errno))</pre> |
| С | eval | <pre>len = %len(%trimr(line))</pre> |
| С | callp | <pre>writeline(STDERR: %addr(line): len) </pre> |
| С | return | -1 |
| С | endif | |
| С | eval | line = fnwithdir |
| c | eval | len = %len(%trimr(line)) |
| c | callp | <pre>writeline(STDOUT: %addr(line): len)</pre> |
| C | Callp | writerine(Siboor. Saddr(Tine). Ten) |
| С | eval | p_statds = %addr(mystat) |
| c | if | S ISDIR(st mode) |
| C | if | show_dir(fnwithdir) < 0 |
| c | return | -1 |
| C | endif | - |
| c | endif | |
| - | 011011 | |
| С | endif | |
| | | |
| С | eval | p_dirent = readdir(dir) |
| С | enddo | |
| | | |

*/

```
c callp closedir(dir)
c return 0
P E
```

```
* This tests a file mode to see if a file is a directory.
 * Here is the C code we're trying to duplicate:
       #define _S_IFDIR 0040000
 *
       #define S_ISDIR(mode) (((mode) & 0370000) == _S_IFDIR)
 *
 * 1) ((mode) & 0370000) takes the file's mode and performs a
       bitwise AND with the octal constant 0370000. In binary,
       that constant looks like: 000000000000001111100000000000
 *
       The effect of this code is to turn off all bits in the
 *
 *
       mode, except those marked with a '1' in the binary bitmask.
 * 2) ((result of #1) == _S_IFDIR) What this does is compare
       the result of step 1, above with the \_S\_IFDIR, which
 *
       is defined to be the octal constant 0040000. In decimal,
 *
       that octal constant is 16384.
 *****
P S ISDIR
                В
D S_ISDIR
                ΡI
                             1 N
D mode
                            10U 0 value
D
                DS
D dirmode
                       1
                             4U 0
                             1 A
D byte1
                       1
D byte2
                       2
                             2A
D byte3
                       3
                             3A
D byte4
                       4
                             4A
C* Turn off bits in the mode, as in step (1) above.
С
                  eval
                          dirmode = mode
                  bitoff
                           x'FF'
                                       byte1
С
                  bitoff
                                       byte2
С
                          x'FE'
                  bitoff
                           x'0F'
                                       byte3
С
                  bitoff
                           x'FF'
                                       byte4
С
C\star Compare the result to 0040000, and return true or false.
С
                  if
                           dirmode = 16384
С
                  return
                           *On
                  else
С
С
                  return
                           *Off
                  endif
С
Ρ
                Е
```

/DEFINE ERRNO_LOAD_PROCEDURE
/COPY IFSEBOOK/QRPGLESRC,ERRNO_H

Once again, you'll need to compile the program, start up QSHELL, and run it from the QSHELL prompt. The command that you'll run will be:

/QSYS.LIB/IFSEBOOK.LIB/CH7RECURSE.PGM